Accompanying file for the paper

On the scaling of random Tamari Intervals and Schnyder woods of random triangulations (with an asymptotic D-finite trick)

Guillaume Chapuy, 2024.

The numbering of sections and equations refer to arxiv v1.

Section 2

Section 2.1: classical case

```
> # Equation from Bousquet-Mélou, Fusy, Préville-Ratelle (already in Chapoton's paper).
  # F=F(t,x) enumerates intervals with t^size x^(contacts of lower path)
  #  f=F(t,1)
  -F+x+x*t*F*(F-f)/(x-1):
  eq0:=numer(%);
```

$$eq0 := F^2\,t\,x - F\,f\,t\,x - F\,x + x^2 + F - x \tag{1.1.1}$$

```
> # That same paper gives a rational parametrization of the solution in new variables (z,u) that reparametrize (t,x)
  eq1:=-t+z*(1-z)^3;
  eq2:=-x+(1+u)/(1+z*u)^2;
  eq3:=-f+(1-2*z)/(1-z)^3;
  eq4:=-F+(1+u)/(1+z*u)/(1-z)^3*(1-2*z-z^2*u);
```

$$eq1 := -t + z\,(1-z)^3$$

$$eq2 := -x + \frac{1+u}{(u\,z+1)^2}$$

$$eq3 := -f + \frac{1-2z}{(1-z)^3}$$

$$eq4 := -F + \frac{(1+u)\left(-u z^2 - 2z + 1\right)}{(u z + 1)(1-z)^3} \quad (1.1.2$$

```
> # One can check that this works (!)
subs(t=solve(eq1,t), x=solve(eq2,x),f=solve(eq3,f),F=solve(eq4,F),eq0);
factor(%);
```

$$-\frac{(1+u)^3 z \left(u z^2+2z-1\right)^2}{(u z+1)^4(-1+z)^3} + \frac{(1+u)^2 z \left(u z^2+2z-1\right)(-1+2z)}{(u z+1)^3(-1+z)^3} - \frac{(1+u)^2\left(u z^2+2z-1\right)}{(u z+1)^3(-1+z)^3} + \frac{(1+u)^2}{(u z+1)^4} + \frac{(1+u)\left(u z^2+2z-1\right)}{(u z+1)(-1+z)^3} - \frac{1+u}{(u z+1)^2}$$

$$0 \quad (1.1.3$$

```
> # By eliminating t we obtain an equation relating f and z, and even solve it:
valf_z:=solve(factor(eliminate({eq0,eq2,eq3},x)[2,1]),f);
# and look at its expansion:
series(%,z=0);

# Note that we can also go back to the variable t to see (for real) the numbers of Tamari
intervals of size n:
devf_z:=subs(z=solve(eq1,z),valf_z):
devf_t:=series(%,t=0,21):
series(%,t,7);

# And in fact we can also see the bivariate expansion which might be useful later on
subs(f=devf_t,solve(eq0,F)[1]):
devF_tx:=simplify(series(%,t=0,12)) assuming x<1:
series(%,t,8);
```

$$valf\_z := \frac{-1+2z}{z^3-3z^2+3z-1}$$

$$1 + z - 2z^3 - 5z^4 - 9z^5 + O\left(z^6\right)$$

$$1 + t + 3t^2 + 13t^3 + 68t^4 + 399t^5 + 2530t^6 + O\left(t^7\right)$$

$$x+x^2 t+(2x+1)x^2t^2+(5x^2+5x+3)x^2t^3+(14x^3+21x^2+20x+13)x^2t^4+(42x^4+84x^3+105x^2+100x+68)x^2t^5 + (132x^5+330x^4+504x^3+595x^2+570x+399)x^2t^6+(429x^6+1287x^5+2310x^4+3192x^3+3675x^2+3542x + 2530)x^2t^7+O(t^8) \quad (1.1.4$$

Section 2.2: height on the upper path

```
> # PROP 2.2: Equation for the generating function H=H(t,x,s)
  # counting intervals with a marked abscissa in [0,2n] and s^{upper height at that abscissa}
  # Here h= H(1) = H(t,1,s), and F, f, are as above.

  eq6:=-H+F+s*x*t*F*(H-h)/(x-1)+x*t*H*(F-f)/(x-1):
  collect(%,H);

  # Kernel of the equation, and remainder once the kernel is cancelled:
  ker1:=subs(x=x,coeff(%,H,1));
  rem1:=subs(x=x,coeff(%%,H,0));
```

$$\left(-1+\frac{s\,x\,t\,F}{x-1}+\frac{x\,t\,(F-f)}{x-1}\right)H+F-\frac{s\,x\,t\,F\,h}{x-1}$$

$$ker1 := -1+\frac{s\,x\,t\,F}{x-1}+\frac{x\,t\,(F-f)}{x-1}$$

$$rem1 := F-\frac{s\,x\,t\,F\,h}{x-1} \quad (1.2.1$$

```
> # We express the kernel and remainder with the variables (u,z)
  # and call U the root, determined by the cancellation of the kernel
  subs(t=solve(eq1,t), x=solve(eq2,x),f=solve(eq3,f),F=solve(eq4,F),ker1):
  numer(factor(subs(u=U,%)));
  eqU:=map(factor, collect(%,s));

  # This equation can also be put in the form s=Rat_s(z,U):
  Rat_s:=factor(solve(%,s));

  # Cancelling the remainder, we solve for h=H(1) in terms of z and this U
  subs(s=Rat_s,t=solve(eq1,t), x=solve(eq2,x),f=solve(eq3,f),F=solve(eq4,F),solve(rem1,h)):
  valueh_zU:=factor(subs(u=U,%));
```

```
# This proves Theorem 2.3.
```

$$U^3 s z^3 + 2 U^2 s z^3 + 2 U^2 s z^2 + U s z^3 - U^2 s z + 4 U s z^2 - U z^3 - 2 U s z + 3 U z^2 + 2 s z^2 - 3 U z - s z + U$$

$$eqU := s (1+U)^2 z \left(U z^2 + 2 z - 1\right) - U z^3 + 3 U z^2 - 3 z U + U$$

$$Rat\_s := \frac{U(-1+z)^3}{z(1+U)^2 \left(U z^2 + 2 z - 1\right)}$$

$$valueh\_zU := \frac{\left(U z^2 + 2 z - 1\right)^2 (1+U)}{(-1+z)^6} \quad \textbf{(1.2.2}$$

```
> # Note that eliminating U we can obtain an equation relating h=H(1) and z:
  final_eq_hzs:=factor(eliminate({s=Rat_s,h=valueh_zU},{U})[2,1]);
```

$$final\_eq\_hzs := h^3 s^3 z^{10} - 9 h^3 s^3 z^9 + 36 h^3 s^3 z^8 - 84 h^3 s^3 z^7 + 126 h^3 s^3 z^6 + 4 h^2 s^2 z^8 - 126 h^3 s^3 z^5 - 26 h^2 s^2 z^7 + 84 h^3 s^3 z^4 + 73 h^2 s^2 z^6 - 36 h^3 s^3 z^3 - 116 h^2 s^2 z^5 - h s^2 z^6 + 9 h^3 s^3 z^2 + 115 h^2 s^2 z^4 + 6 h s^2 z^5 + 2 h s z^6 - h^3 s^3 z - 74 h^2 s^2 z^3 - 15 h s^2 z^4 - 4 h s z^5 - h z^6 + 31 h^2 s^2 z^2 + 20 h s^2 z^3 - h s z^4 + 6 h z^5 - 8 h^2 s^2 z - 15 h s^2 z^2 + 7 h s z^3 - 15 h z^4 + h^2 s^2 + 6 h s^2 z - 5 h s z^2 + 20 h z^3 - h s^2 + h s z - 15 h z^2 + 6 h z + 4 z^2 - h - 4 z + 1 \quad \textbf{(1.2.3}$$

```
> # From the obtained equation we can check the expansion of H(1):
  eliminate({final_eq_hzs,eq1},z)[2,1]:
  devh_t:=series(RootOf(%,h),t=0,7);
```

$$devh\_t := 1 + (s+2) t + \left(2 s^2 + 6 s + 7\right) t^2 + \left(5 s^3 + 20 s^2 + 34 s + 32\right) t^3 + \left(14 s^4 + 70 s^3 + 155 s^2 + 202 s + 171\right) t^4 + \left(42 s^5 + 252 s^4 + 686 s^3 + 1130 s^2 + 1267 s + 1012\right) t^5 + \left(132 s^6 + 924 s^5 + 2982 s^4 + 5922 s^3 + 8161 s^2 + 8334 s + 6435\right) t^6 + \mathrm{O}\left(t^7\right) \quad \textbf{(1.2.4}$$

```
> # ... and we can also check the expansion of H(x):
  solve(eq6,H):
  subs(F=devF_tx, f=devf_t,h=devh_t,%):
  factor(series(%,t=0,4));
```

$$x + (s+2) x^2 t + \left(s^2 x + s^2 + 4 s x + 2 s + 5 x + 2\right) x^2 t^2 + \left(s^3 x^2 + 2 s^3 x + 6 s^2 x^2 + 2 s^3 + 8 s^2 x + 14 s x^2 + 6 s^2 + 13 s x + 14 x^2 + 7 s + 12 x + 6\right) x^2 t^3 + \mathrm{O}\left(t^4\right) \quad \textbf{(1.2.5}$$

Section 3: lower path

Section 3.1: writing the equation

```
> ## We consider again intervals (Pn,Qn) with a marked abscissa i
  # and take the weights
  # x^(contacts before i), y^(contacts after i) w^(P(i)) t^size
  # As in the paper we call G(x,y) the corresponding function.
  # In equations below we note
  #          G  =  G(x,y)
  #        Fx = F(x)    (the series of unmarked intervals)
  #        Fy = F(x=y) (the series of unmarked intervals, but changing x for y)
  #         g1 = G(1,y) (also depends on w)
  #         g11 = G(1,1)    (also depends on w)
  #          f = F(1) (from previous sections)

  # The combinatorial decomposition of intervals gives (PROP 3.1):
  eqq0:=
   -G + t*x*(g1-g11)*Fy*w/(y-1)+t*x^2*(G-Fx*y/x+f*y-g1)*Fy/(y*(x-1))+x*t*(Fx-f)*G/(x-1)+x*(t*
  y^2*(Fy-f)/(y-1)-t*y^2*f)*Fy/y^2+Fy;
```

$$eqq0 := -G + \frac{t\,x\,(g1-g11)\,Fy\,w}{y-1} + \frac{t\,x^2\left(G-\frac{Fx\,y}{x}+f\,y-g1\right)Fy}{y\,(x-1)} + \frac{x\,t\,(Fx-f)\,G}{x-1} + \frac{x\left(\frac{t\,y^2\,(Fy-f)}{y-1}-t\,y^2 f\right)Fy}{y^2} + Fy \qquad \text{(2.1.1}$$

```
> # First we compute the first few terms of G, by iterating the equation:
  devG:=0:
  for i from 0 to 10 do
   G+eqq0:
   subs(f=devf_t,Fx=devF_tx,Fy=subs(x=y,devF_tx),%);
   subs(G=devG,g1=subs(x=1,devG),g11=subs(x=1,y=1,devG),%);
   devG:= simplify(factor(series(%,t,i+1))):
  od:
  devG:

  # we check that for x=y=1 we indeed find the known series
  expand(series(devG,t=0,3));
  collect(coeff(%,t,2),{x,y});
  expand(subs(x=1,y=1,w=1,devG));
  series(2*t*diff(devf_t,t)+devf_t,t=0);
```

```
# Here are the coefficients of t^n for n<=2, one can check that this is correct by drawing all the cases (!)
expand(series(devG,t=0,6));
```

$$y+\left(w x y+x y+y^2\right) t+\left(w^2 x y+2 w x^2 y+2 w x y^2+2 w x y+2 x^2 y+2 x y^2+2 y^3+x y+y^2\right) t^2+\mathrm{O}\left(t^3\right)$$

$$(2 w+2) y x^2+\left((2 w+2) y^2+\left(w^2+2 w+1\right) y\right) x+2 y^3+y^2$$

$$1+3 t+15 t^2+91 t^3+612 t^4+4389 t^5+32890 t^6+254475 t^7+2017356 t^8+16301164 t^9+133767543 t^{10}+\mathrm{O}\left(t^{11}\right)$$

$$1+3 t+15 t^2+91 t^3+612 t^4+4389 t^5+\mathrm{O}\left(t^6\right)$$

$$\begin{aligned} & y+\left(w x y+x y+y^2\right) t+\left(w^2 x y+2 w x^2 y+2 w x y^2+2 w x y+2 x^2 y+2 x y^2+2 y^3+x y+y^2\right) t^2+\left(w^3 x y+2 w^2 x^2 y\right. \\ & +3 w^2 x y^2+5 w x^3 y+5 w x^2 y^2+5 w x y^3+6 w^2 x y+7 w x^2 y+8 w x y^2+5 x^3 y+5 x^2 y^2+5 x y^3+5 y^4+8 w x y+5 x^2 y \\ & \left.+5 x y^2+5 y^3+3 x y+3 y^2\right) t^3+\left(w^4 x y+2 w^3 x^2 y+4 w^3 x y^2+5 w^2 x^3 y+7 w^2 x^2 y^2+9 w^2 x y^3+14 w x^4 y+14 w x^3 y^2\right. \\ & +14 w x^2 y^3+14 w x y^4+11 w^3 x y+15 w^2 x^2 y+25 w^2 x y^2+26 w x^3 y+28 w x^2 y^2+30 w x y^3+14 x^4 y+14 x^3 y^2+14 x^2 y^3 \\ & +14 x y^4+14 y^5+38 w^2 x y+33 w x^2 y+41 w x y^2+21 x^3 y+21 x^2 y^2+21 x y^3+21 y^4+41 w x y+20 x^2 y+20 x y^2+20 y^3 \\ & \left.+13 x y+13 y^2\right) t^4+\left(w^5 x y+2 w^4 x^2 y+5 w^4 x y^2+5 w^3 x^3 y+9 w^3 x^2 y^2+14 w^3 x y^3+14 w^2 x^4 y+19 w^2 x^3 y^2+23 w^2 x^2 y^3\right. \\ & +28 w^2 x y^4+42 w x^5 y+42 w x^4 y^2+42 w x^3 y^3+42 w x^2 y^4+42 w x y^5+17 w^4 x y+25 w^3 x^2 y+53 w^3 x y^2+46 w^2 x^3 y \\ & +69 w^2 x^2 y^2+95 w^2 x y^3+98 w x^4 y+103 w x^3 y^2+107 w x^2 y^3+112 w x y^4+42 x^5 y+42 x^4 y^2+42 x^3 y^3+42 x^2 y^4+42 x y^5 \\ & +42 y^6+99 w^3 x y+105 w^2 x^2 y+188 w^2 x y^2+146 w x^3 y+165 w x^2 y^2+186 w x y^3+84 x^4 y+84 x^3 y^2+84 x^2 y^3+84 x y^4 \\ & +84 y^5+255 w^2 x y+182 w x^2 y+240 w x y^2+105 x^3 y+105 x^2 y^2+105 x y^3+105 y^4+240 w x y+100 x^2 y+100 x y^2 \\ & \left.+100 y^3+68 x y+68 y^2\right) t^5+\mathrm{O}\left(t^6\right) \end{aligned} \tag{2.1.2}$$

Section 3.2: solution

First step: eliminate x (or u)

```
> # Our equation has now two catalytic variables but it is still linear, let us look at the kernel:
```

```
collect(eqq0,G):
# bivariate kernel
ker2:=coeff(%,G,1);
# and remainder term
rem2:=coeff(%%,G,0);

# Since everything with one variable is known from previous works, we can express the kernel in the new variables
# Here we introduce a new variable v which is to y what u is to x.
subs(F=Fx,u=u,{eq1,eq2,eq3,eq4}):
subs(F=Fy,u=v,x=y,{eq1,eq2,eq3,eq4}):
sys2:=% union %%:
eliminate(sys2 union {KK=ker2},{Fx,Fy,x,y,f,t})[2,1]:

# Here is the kernel (more precisely its numerator) in the new variables!
factor(numer(factor(solve(%,KK)))):
ker2N:=map(x->collect(x,u),%);
```

$$ker2 := -1 + \frac{t\,x^2\,Fy}{y\,(x-1)} + \frac{x\,t\,(Fx-f)}{x-1}$$

$$rem2 := \frac{t\,x\,(g1-g11)\,Fy\,w}{y-1} + \frac{t\,x^2\left(-\frac{Fx\,y}{x}+f\,y-g1\right)Fy}{y\,(x-1)} + \frac{x\left(\frac{t\,y^2\,(Fy-f)}{y-1}-t\,y^2\,f\right)Fy}{y^2} + Fy$$

$$ker2N := \left(\left(v\,z^2+z^2\right)u+v\,z^2+2\,z-1\right)\left(\left(v\,z^2-z^2+3\,z-1\right)u+v\,z^2+z\right) \quad \textbf{(2.2.1.1}$$

```
> # Now let us look at the two roots root U=u(z,v,w) of the kernel
solve(ker2N,u):
valU:=[%][2];
otherU:=[%%][1]:
# One of the roots is a non-singular power series in z (it is called U_0 in the paper)
devU:=simplify(series(valU,z=0,5));
# Note that the other root if NOT a power series
simplify(series(otherU,z=0,3));
```

$$valU := -\frac{z\,(v\,z+1)}{v\,z^2-z^2+3\,z-1}$$

$$devU := z + (v+3)\, z^2 + (4\, v + 8)\, z^3 + \left(v^2 + 14\, v + 21\right) z^4 + \mathrm{O}\left(z^5\right)$$

$$\frac{\frac{1}{1+v}}{z^2} - \frac{\frac{2}{1+v}}{z} - \frac{v}{1+v} \quad (2.2.1.2$$

```
> # Now we will look at the cancellation of the remainder term
  # We first express this remainder in the new variables
  factor((subs(t=solve(eq1,t),f=solve(eq3,f),rem2))):
  subs(y=subs(u=v,solve(eq2,x)),%):
  simplify(subs(Fx=solve(eq4,F),Fy=subs(u=v,solve(eq4,F)),%)):
  rem2N:=factor(subs(x=solve(eq2,x),%)*v*(v*z+1)*(-1+z)^3*(u*z^2+2*z-1)*u*(u*z+1)^4/(1+u*z)
  ^2):

  # This quantity is explicit, it has 228 terms:
  nops(%);
```

$$228 \quad (2.2.1.3$$

```
> # This is the equation we obtain, Eq (30) in the paper. We have successfully eliminated G
  and the variable u:
  subs(u=valU, rem2N):
  eqL:=factor(numer(%)/z/(z-1)^4/(1+v*z));

  # The equation has degree 1 in G_1(v) and 91 terms:
  degree(%,g1);
  nops(eqL);
```

$$eqL := -g1\, v^4\, w\, z^8 + g11\, v^4\, w\, z^8 + 3\, g1\, v^4\, w\, z^7 - g1\, v^4\, z^8 - g1\, v^3\, w\, z^8 - 3\, g11\, v^4\, w\, z^7 + g11\, v^3\, w\, z^8 - 3\, g1\, v^4\, w\, z^6 + 3\, g1\, v^4\, z^7$$
$$- 2\, g1\, v^3\, w\, z^7 + g1\, v^3\, z^8 + 3\, g11\, v^4\, w\, z^6 + 2\, g11\, v^3\, w\, z^7 + g1\, v^4\, w\, z^5 - 3\, g1\, v^4\, z^6 + 13\, g1\, v^3\, w\, z^6 - 9\, g1\, v^3\, z^7 - 5\, g1\, v^2\, w\, z^7$$
$$- g11\, v^4\, w\, z^5 - 13\, g11\, v^3\, w\, z^6 + 5\, g11\, v^2\, w\, z^7 + g1\, v^4\, z^5 - 17\, g1\, v^3\, w\, z^5 + 23\, g1\, v^3\, z^6 + 9\, g1\, v^2\, w\, z^6 + 3\, g1\, v^2\, z^7$$
$$+ 17\, g11\, v^3\, w\, z^5 - 9\, g11\, v^2\, w\, z^6 + 8\, g1\, v^3\, w\, z^4 - 25\, g1\, v^3\, z^5 + 5\, g1\, v^2\, w\, z^5 - 21\, g1\, v^2\, z^6 - 7\, g1\, v\, w\, z^6 - 8\, g11\, v^3\, w\, z^4$$
$$- 5\, g11\, v^2\, w\, z^5 + 7\, g11\, v\, w\, z^6 - g1\, v^3\, w\, z^3 + 12\, g1\, v^3\, z^4 - 19\, g1\, v^2\, w\, z^4 + 52\, g1\, v^2\, z^5 + 20\, g1\, v\, w\, z^5 + 2\, g1\, v\, z^6 + g11\, v^3\, w\, z^3$$
$$+ 19\, g11\, v^2\, w\, z^4 - 20\, g11\, v\, w\, z^5 - v^4\, z^4 - 2\, g1\, v^3\, z^3 + 12\, g1\, v^2\, w\, z^3 - 61\, g1\, v^2\, z^4 - 17\, g1\, v\, w\, z^4 - 13\, g1\, v\, z^5 - 3\, g1\, w\, z^5$$
$$- 12\, g11\, v^2\, w\, z^3 + 17\, g11\, v\, w\, z^4 + 3\, g11\, w\, z^5 + v^3\, z^4 - 2\, g1\, v^2\, w\, z^2 + 36\, g1\, v^2\, z^3 + g1\, v\, w\, z^3 + 32\, g1\, v\, z^4 + 10\, g1\, w\, z^4$$
$$+ 2\, g11\, v^2\, w\, z^2 - g11\, v\, w\, z^3 - 10\, g11\, w\, z^4 - 6\, v^3\, z^3 + v^2\, z^4 - 10\, g1\, v^2\, z^2 + 4\, g1\, v\, w\, z^2 - 39\, g1\, v\, z^3 - 12\, g1\, w\, z^3 - 4\, g11\, v\, w\, z^2$$
$$+ 12\, g11\, w\, z^3 + 2\, v^3\, z^2 - v\, z^4 + g1\, v^2\, z - g1\, v\, w\, z + 25\, g1\, v\, z^2 + 6\, g1\, w\, z^2 + g11\, v\, w\, z - 6\, g11\, w\, z^2 - 9\, v^2\, z^2 + 6\, v\, z^3$$

$$-8\,g1\,v\,z-g1\,w\,z+g11\,w\,z+6\,v^2 z-11\,v\,z^2+g1\,v-v^2+6\,v\,z-v$$
$$1$$
$$91 \qquad (2.2.1.4$$

Second step: eliminate y (or v)

```
> # The equation we obtained on g1 is linear, it has two coefficients.
  aa:=coeff(eqL,g1,1);
  bb:=coeff(eqL,g1,0);
```

$$aa := -v^4 w z^8+3 v^4 w z^7-v^4 z^8-v^3 w z^8-3 v^4 w z^6+3 v^4 z^7-2 v^3 w z^7+v^3 z^8+v^4 w z^5-3 v^4 z^6+13 v^3 w z^6-9 v^3 z^7-5 v^2 w z^7+v^4 z^5-17 v^3 w z^5+23 v^3 z^6+9 v^2 w z^6+3 v^2 z^7+8 v^3 w z^4-25 v^3 z^5+5 v^2 w z^5-21 v^2 z^6-7 v w z^6-v^3 w z^3+12 v^3 z^4-19 v^2 w z^4+52 v^2 z^5+20 v w z^5+2 v z^6-2 v^3 z^3+12 v^2 w z^3-61 v^2 z^4-17 v w z^4-13 v z^5-3 w z^5-2 v^2 w z^2+36 v^2 z^3+v w z^3+32 v z^4+10 w z^4-10 v^2 z^2+4 v w z^2-39 v z^3-12 w z^3+v^2 z-v w z+25 v z^2+6 w z^2-8 v z-w z+v$$

$$bb := g11\,v^4 w z^8-3\,g11\,v^4 w z^7+g11\,v^3 w z^8+3\,g11\,v^4 w z^6+2\,g11\,v^3 w z^7-g11\,v^4 w z^5-13\,g11\,v^3 w z^6+5\,g11\,v^2 w z^7+17\,g11\,v^3 w z^5-9\,g11\,v^2 w z^6-8\,g11\,v^3 w z^4-5\,g11\,v^2 w z^5+7\,g11\,v\,w z^6+g11\,v^3 w z^3+19\,g11\,v^2 w z^4-20\,g11\,v\,w z^5-v^4 z^4-12\,g11\,v^2 w z^3+17\,g11\,v\,w z^4+3\,g11\,w z^5+v^3 z^4+2\,g11\,v^2 w z^2-g11\,v\,w z^3-10\,g11\,w z^4-6\,v^3 z^3+v^2 z^4-4\,g11\,v\,w z^2+12\,g11\,w z^3+2\,v^3 z^2-v z^4+g11\,v\,w z-6\,g11\,w z^2-9\,v^2 z^2+6\,v z^3+g11\,w z+6\,v^2 z-11\,v z^2-v^2+6\,v z-v \qquad (2.2.2.1$$

```
> # Let us look at the v-roots of the linear coefficient:
  algcurves[puiseux](RootOf(aa,v),z=0,2):
  map(x-> collect(x,z),%);
  # One of them is a power series in z
```

$$\left\{w\,z,\ -\frac{1}{z},\ -\frac{z}{w}-\frac{3}{z}+\frac{1}{z^2},\ \frac{(-w^2+1)\,z}{w}+\frac{-w+1}{w+1}+\frac{-w-2}{(w+1)\,z}+\frac{1}{(w+1)\,z^2}\right\} \qquad (2.2.2.2$$

```
> # By substituting v=(that root) both coefficients vanish.
  # We can thus eliminate v beween these two equations and we get our final equation for
  g11:
  factor(eliminate({aa,bb},v)[2,1])/(w*z*(3*z-1)*(-1+2*z)*(-1+z));
  finaleq_g11:=map(factor,collect(%,g11));

  # This is the equation given in Theorem 3.2 !
```

$$g11^3 w z^{10} - 9 g11^3 w z^9 + 36 g11^3 w z^8 + 2 g11^2 w^2 z^8 - 84 g11^3 w z^7 - 13 g11^2 w^2 z^7 + 126 g11^3 w z^6 + 36 g11^2 w^2 z^6 + 2 g11^2 z^8 - 126 g11^3 w z^5 - 55 g11^2 w^2 z^5 + g11^2 w z^6 - 13 g11^2 z^7 - g11 w^2 z^6 + 84 g11^3 w z^4 + 50 g11^2 w^2 z^4 - 6 g11^2 w z^5 + 36 g11^2 z^6 + 6 g11 w^2 z^5 + 2 g11 w z^6 - 36 g11^3 w z^3 - 27 g11^2 w^2 z^3 + 15 g11^2 w z^4 - 55 g11^2 z^5 - 13 g11 w^2 z^4 - 4 g11 w z^5 - g11 z^6 + 9 g11^3 w z^2 + 8 g11^2 w^2 z^2 - 20 g11^2 w z^3 + 50 g11^2 z^4 + 13 g11 w^2 z^3 - 5 g11 w z^4 + 6 g11 z^5 - g11^3 w z - g11^2 w^2 z + 15 g11^2 w z^2 - 27 g11^2 z^3 - 6 g11 w^2 z^2 + 21 g11 w z^3 - 13 g11 z^4 - 6 g11^2 w z + 8 g11^2 z^2 + g11 w^2 z - 23 g11 w z^2 + 13 g11 z^3 + g11^2 w - g11^2 z + 11 g11 w z - 6 g11 z^2 + 4 w z^2 - 2 g11 w + g11 z - 4 w z + w$$

$$finaleq\_g11 := w z (-1+z)^9 g11^3 + (-1+z)^6 (2 w^2 z^2 - w^2 z + 2 z^2 + w - z) g11^2 - (-1+z)^3 (w^2 z^3 - 3 w^2 z^2 - 2 w z^3 + w^2 z - 2 w z^2 + z^3 + 5 w z - 3 z^2 - 2 w + z) g11 + 4 w z^2 - 4 w z + w \quad (2.2.2.3)$$

```
> # Note than one can check the expansion against the one computed from the functional
  equation:
  subs(t=solve(eq1,t),devG):
  series(%,z=0,10):
  devG11z:=factor(subs(x=1,y=1,%));

  series(RootOf(finaleq_g11,g11),z=0,10):
  factor(subs(RootOf(_Z^2*w-_Z*w^2-4*_Z*w+2*w^2-_Z+5*w+2)=w+2,%)):
  series(%-devG11z,z=0,10);
```

$$devG11z := 1 + (w+2) z + (w+2)(w+1) z^2 + (w^3 + 5 w^2 + 5 w - 1) z^3 + (w^4 + 8 w^3 + 15 w^2 + 2 w - 11) z^4 + (w^5 + 12 w^4 + 37 w^3 + 30 w^2 - 23 w - 36) z^5 + (w^6 + 17 w^5 + 78 w^4 + 123 w^3 + 20 w^2 - 119 w - 92) z^6 + (w^7 + 23 w^6 + 147 w^5 + 360 w^4 + 294 w^3 - 164 w^2 - 414 w - 211) z^7 + (w^8 + 30 w^7 + 255 w^6 + 879 w^5 + 1276 w^4 + 346 w^3 - 1066 w^2 - 1219 w - 457) z^8 + (w^9 + 38 w^8 + 415 w^7 + 1902 w^6 + 4053 w^5 + 3410 w^4 - 1181 w^3 - 4382 w^2 - 3243 w - 958) z^9 + O(z^{10})$$

$$0 \quad (2.2.2.4)$$

```
> # And here is the equation in the variable t (not written in the paper)
  finaleq_g11t:=factor(eliminate({finaleq_g11,eq1},z)[2,1]);
```

$$finaleq\_g11t := g11^{12} t^{10} w^4 - 3 g11^{11} t^9 w^5 + 8 g11^{10} t^9 w^6 - 4 g11^{11} t^9 w^4 + 4 g11^{10} t^9 w^5 + 3 g11^{10} t^8 w^6 + 20 g11^9 t^8 w^7 + 16 g11^8 t^8 w^8 - 3 g11^{11} t^9 w^3 + 8 g11^{10} t^9 w^4 + 12 g11^{10} t^8 w^5 - 41 g11^9 t^8 w^6 - g11^9 t^7 w^7 + 16 g11^8 t^8 w^7 - g11^8 t^7 w^8 + 4 g11^{10} t^9 w^3 + 15 g11^{10} t^8 w^4 - 54 g11^9 t^8 w^5 - 12 g11^9 t^7 w^6 + 38 g11^8 t^8 w^6 - 74 g11^8 t^7 w^7 - 44 g11^7 t^7 w^8 + 8 g11^{10} t^9 w^2 + 12 g11^{10} t^8 w^3 - 42 g11^9 t^8 w^4 - 27 g11^9 t^7 w^5 + 24 g11^8 t^8 w^5 + 27 g11^8 t^7 w^6 + 4 g11^8 t^6 w^7 - 65 g11^7 t^7 w^7 + 3 g11^7 t^6 w^8 \quad (2.2.2.5)$$

$$+ 8\, g11^{6} t^{7} w^{8} + 3\, g11^{10} t^{8} w^{2} - 54\, g11^{9} t^{8} w^{3} - 40\, g11^{9} t^{7} w^{4} + 68\, g11^{8} t^{8} w^{4} + 189\, g11^{8} t^{7} w^{5} + 21\, g11^{8} t^{6} w^{6}$$

$$- 232\, g11^{7} t^{7} w^{6} + 103\, g11^{7} t^{6} w^{7} - 28\, g11^{6} t^{7} w^{7} + 39\, g11^{6} t^{6} w^{8} - 41\, g11^{9} t^{8} w^{2} - 27\, g11^{9} t^{7} w^{3} + 24\, g11^{8} t^{8} w^{3}$$

$$+ 168\, g11^{8} t^{7} w^{4} + 48\, g11^{8} t^{6} w^{5} - 167\, g11^{7} t^{7} w^{5} + 129\, g11^{7} t^{6} w^{6} - 6\, g11^{7} t^{5} w^{7} + 40\, g11^{6} t^{7} w^{6} + 94\, g11^{6} t^{6} w^{7}$$

$$- 3\, g11^{6} t^{5} w^{8} - 19\, g11^{5} t^{6} w^{8} + 20\, g11^{9} t^{8} w - 12\, g11^{9} t^{7} w^{2} + 38\, g11^{8} t^{8} w^{2} + 189\, g11^{8} t^{7} w^{3} + 64\, g11^{8} t^{6} w^{4} - 520\, g11^{7} t^{7} w^{4}$$

$$- 261\, g11^{7} t^{6} w^{5} - 24\, g11^{7} t^{5} w^{6} - 36\, g11^{6} t^{7} w^{5} + 399\, g11^{6} t^{6} w^{6} - 65\, g11^{6} t^{5} w^{7} + 22\, g11^{5} t^{6} w^{7} - 10\, g11^{5} t^{5} w^{8} + g11^{4} t^{6} w^{8}$$

$$- g11^{9} t^{7} w + 16\, g11^{8} t^{8} w + 27\, g11^{8} t^{7} w^{2} + 48\, g11^{8} t^{6} w^{3} - 167\, g11^{7} t^{7} w^{3} - 410\, g11^{7} t^{6} w^{4} - 60\, g11^{7} t^{5} w^{5} + 32\, g11^{6} t^{7} w^{4}$$

$$+ 394\, g11^{6} t^{6} w^{5} - 256\, g11^{6} t^{5} w^{6} + 4\, g11^{6} t^{4} w^{7} - 264\, g11^{5} t^{6} w^{6} - 55\, g11^{5} t^{5} w^{7} + g11^{5} t^{4} w^{8} - 8\, g11^{4} t^{6} w^{7} + 11\, g11^{4} t^{5} w^{8}$$

$$+ 16\, g11^{8} t^{8} - 74\, g11^{8} t^{7} w + 21\, g11^{8} t^{6} w^{2} - 232\, g11^{7} t^{7} w^{2} - 261\, g11^{7} t^{6} w^{3} - 72\, g11^{7} t^{5} w^{4} - 36\, g11^{6} t^{7} w^{3}$$

$$+ 2052\, g11^{6} t^{6} w^{4} + 92\, g11^{6} t^{5} w^{5} + 21\, g11^{6} t^{4} w^{6} + 394\, g11^{5} t^{6} w^{5} - 203\, g11^{5} t^{5} w^{6} + 17\, g11^{5} t^{4} w^{7} + 28\, g11^{4} t^{6} w^{6}$$

$$+ 29\, g11^{4} t^{5} w^{7} - g11^{4} t^{4} w^{8} - g11^{8} t^{7} + 4\, g11^{8} t^{6} w - 65\, g11^{7} t^{7} w + 129\, g11^{7} t^{6} w^{2} - 60\, g11^{7} t^{5} w^{3} + 40\, g11^{6} t^{7} w^{2}$$

$$+ 394\, g11^{6} t^{6} w^{3} + 506\, g11^{6} t^{5} w^{4} + 48\, g11^{6} t^{4} w^{5} - 266\, g11^{5} t^{6} w^{4} - 379\, g11^{5} t^{5} w^{5} + 171\, g11^{5} t^{4} w^{6} - g11^{5} t^{3} w^{7}$$

$$- 56\, g11^{4} t^{6} w^{5} + 392\, g11^{4} t^{5} w^{6} + 8\, g11^{4} t^{4} w^{7} + 16\, g11^{3} t^{5} w^{7} - 44\, g11^{7} t^{7} + 103\, g11^{7} t^{6} w - 24\, g11^{7} t^{5} w^{2} - 28\, g11^{6} t^{7} w$$

$$+ 399\, g11^{6} t^{6} w^{2} + 92\, g11^{6} t^{5} w^{3} + 64\, g11^{6} t^{4} w^{4} + 394\, g11^{5} t^{6} w^{3} - 4018\, g11^{5} t^{5} w^{4} + 121\, g11^{5} t^{4} w^{5} - 12\, g11^{5} t^{3} w^{6}$$

$$+ 70\, g11^{4} t^{6} w^{4} - 917\, g11^{4} t^{5} w^{5} - 53\, g11^{4} t^{4} w^{6} - g11^{4} t^{3} w^{7} - 96\, g11^{3} t^{5} w^{6} - 23\, g11^{3} t^{4} w^{7} + 3\, g11^{7} t^{6} - 6\, g11^{7} t^{5} w$$

$$+ 8\, g11^{6} t^{7} + 94\, g11^{6} t^{6} w - 256\, g11^{6} t^{5} w^{2} + 48\, g11^{6} t^{4} w^{3} - 264\, g11^{5} t^{6} w^{2} - 379\, g11^{5} t^{5} w^{3} - 242\, g11^{5} t^{4} w^{4} - 27\, g11^{5} t^{3} w^{5}$$

$$- 56\, g11^{4} t^{6} w^{3} + 1482\, g11^{4} t^{5} w^{4} + 88\, g11^{4} t^{4} w^{5} - 35\, g11^{4} t^{3} w^{6} + 240\, g11^{3} t^{5} w^{5} - 158\, g11^{3} t^{4} w^{6} + 2\, g11^{3} t^{3} w^{7} + 39\, g11^{6} t^{6}$$

$$- 65\, g11^{6} t^{5} w + 21\, g11^{6} t^{4} w^{2} + 22\, g11^{5} t^{6} w - 203\, g11^{5} t^{5} w^{2} + 121\, g11^{5} t^{4} w^{3} - 40\, g11^{5} t^{3} w^{4} + 28\, g11^{4} t^{6} w^{2}$$

$$- 917\, g11^{4} t^{5} w^{3} + 3885\, g11^{4} t^{4} w^{4} - 118\, g11^{4} t^{3} w^{5} + 3\, g11^{4} t^{2} w^{6} - 320\, g11^{3} t^{5} w^{4} + 823\, g11^{3} t^{4} w^{5} + 51\, g11^{3} t^{3} w^{6}$$

$$+ 96\, g11^{2} t^{4} w^{6} - 3\, g11^{6} t^{5} + 4\, g11^{6} t^{4} w - 19\, g11^{5} t^{6} - 55\, g11^{5} t^{5} w + 171\, g11^{5} t^{4} w^{2} - 27\, g11^{5} t^{3} w^{3} - 8\, g11^{4} t^{6} w$$

$$+ 392\, g11^{4} t^{5} w^{2} + 88\, g11^{4} t^{4} w^{3} - 70\, g11^{4} t^{3} w^{4} + 12\, g11^{4} t^{2} w^{5} + 240\, g11^{3} t^{5} w^{3} - 2820\, g11^{3} t^{4} w^{4} + 74\, g11^{3} t^{3} w^{5}$$

$$- 3\, g11^{3} t^{2} w^{6} - 384\, g11^{2} t^{4} w^{5} - 10\, g11^{2} t^{3} w^{6} - 10\, g11^{5} t^{5} + 17\, g11^{5} t^{4} w - 12\, g11^{5} t^{3} w^{2} + g11^{4} t^{6} + 29\, g11^{4} t^{5} w$$

$$- 53\, g11^{4} t^{4} w^{2} - 118\, g11^{4} t^{3} w^{3} + 15\, g11^{4} t^{2} w^{4} - 96\, g11^{3} t^{5} w^{2} + 823\, g11^{3} t^{4} w^{3} - 1666\, g11^{3} t^{3} w^{4} + 22\, g11^{3} t^{2} w^{5}$$

$$+ 576\, g11^{2} t^{4} w^{4} - 264\, g11^{2} t^{3} w^{5} + g11^{5} t^{4} - g11^{5} t^{3} w + 11\, g11^{4} t^{5} + 8\, g11^{4} t^{4} w - 35\, g11^{4} t^{3} w^{2} + 12\, g11^{4} t^{2} w^{3}$$

$$+ 16\, g11^{3} t^{5} w - 158\, g11^{3} t^{4} w^{2} + 74\, g11^{3} t^{3} w^{3} + 112\, g11^{3} t^{2} w^{4} - 3\, g11^{3} t\, w^{5} - 384\, g11^{2} t^{4} w^{3} + 2116\, g11^{2} t^{3} w^{4}$$

$$- 32\, g11^{2} t^{2} w^{5} + 256\, g11\, t^{3} w^{5} - g11^{4} t^{4} - g11^{4} t^{3} w + 3\, g11^{4} t^{2} w^{2} - 23\, g11^{3} t^{4} w + 51\, g11^{3} t^{3} w^{2} + 22\, g11^{3} t^{2} w^{3}$$

$$-4\,g11^3\,t\,w^4+96\,g11^2\,t^4\,w^2-264\,g11^2\,t^3\,w^3+134\,g11^2\,t^2\,w^4+5\,g11^2\,t\,w^5-512\,g11\,t^3\,w^4+2\,g11^3\,t^3\,w-3\,g11^3\,t^2\,w^2$$
$$-3\,g11^3\,t\,w^3-10\,g11^2\,t^3\,w^2-32\,g11^2\,t^2\,w^3-28\,g11^2\,t\,w^4+256\,g11\,t^3\,w^3-512\,g11\,t^2\,w^4-2\,g11\,t\,w^5+5\,g11^2\,t\,w^3$$
$$+g11^2\,w^4+64\,g11\,t\,w^4+256\,t^2\,w^4-2\,g11\,t\,w^3-2\,g11\,w^4-32\,t\,w^4+w^4$$

```
> # Again we can get the expansion:
series(RootOf(finaleq_g11t,g11),t=0,8):
devG11t:=factor(subs(RootOf(_Z^2*w-_Z*w^2-4*_Z*w+2*w^2-_Z+5*w+2)=w+2,%));


# This can be tested against the following value obtained in another worksheet with an independent method (exhaustive generation of intervals!)

1+(w+2)*t+(w^2+6*w+8)*t^2+(w^3+11*w^2+38*w+41)*t^3+(w^4+17*w^3+99*w^2+255*w+240)*t^4+
(w^5+24*w^4+205*w^3+842*w^2+1789*w+1528)*t^5+(w^6+32*w^5+372*w^4+2160*w^3+7025*w^2+12988*
w+10312)*t^6+(w^7+41*w^6+618*w^5+4756*w^4+21263*w^3+58276*w^2+96873*w+72647)*t^7:

simplify(series(%-%%,t=0,8));
```

$$devG11t := 1+(w+2)\,t+(w+4)\,(w+2)\,t^2+\left(w^3+11\,w^2+38\,w+41\right)t^3+\left(w^4+17\,w^3+99\,w^2+255\,w+240\right)t^4+\left(w^5+24\,w^4+205\,w^3+842\,w^2+1789\,w+1528\right)t^5+\left(w^6+32\,w^5+372\,w^4+2160\,w^3+7025\,w^2+12988\,w+10312\right)t^6+\left(w^7+41\,w^6+618\,w^5+4756\,w^4+21263\,w^3+58276\,w^2+96873\,w+72647\right)t^7+O\left(t^8\right)$$

$$O\left(t^8\right) \quad (2.2.2.6$$

Section 4: Tracking up steps on both path

Writing and solving the equation

```
> ### Generating function J(x,y) of intervals (P,Q) of size n with a marked j\in [n], where
### r and s mark the height of the j-th upstep of P and Q respectively
### As before t marks the size and x, y mark contacts before/after the marked step.
### This is proposition 4.1
```

```
> eqq2:=
 -J+(Fy-y)*x/y  +  r*s*t*x*(j1-j11)*Fy/(y-1)  +  s*t*x^2*(J-j1)*Fy/(y*(x-1))+x*t*(Fx-f)*J/
(x-1):
```

```
> # First we compute the first few terms of G, by iterating the equation:
devJ:=0:
for i from 0 to 10 do
 J+eqq2:
 subs(f=devf_t,Fx=devF_tx,Fy=subs(x=y,devF_tx),%);
 subs(J=devJ,j1=subs(x=1,devJ),j11=subs(x=1,y=1,devJ),%);
 devJ:= simplify(factor(series(%,t,i+1))):
od:
devJ:

# we check that for x=y=1 we indeed find the known series
expand(subs(x=1,y=1,r=1,s=1,devJ));
series(t*diff(devf_t,t)-%,t=0);
```

$$t+6\,t^2+39\,t^3+272\,t^4+1995\,t^5+15180\,t^6+118755\,t^7+949344\,t^8+7721604\,t^9+63698830\,t^{10}+\mathrm{O}(t^{11})$$

$$\mathrm{O}(t^6) \qquad \textbf{(3.1.1}$$

```
> # Value obtained by exhaustive generation in an independent worksheet:
vJexh:=x*t*y+(r*s*x*y+s*x^2*y+x^2*y+2*x*y^2+x*y)*t^2+(r^2*s^2*x*y+r*s^2*x^2*y+s^2*x^3*y+r*
s^2*x*y+r*s*x^2*y+3*r*s*x*y^2+s^2*x^2*y+2*s*x^3*y+3*s*x^2*y^2+4*r*s*x*y+2*s*x^2*y+2*x^3*y+2*
x^2*y^2+5*x*y^3+2*x^2*y+5*x*y^2+3*x*y)*t^3+(r^3*s^3*x*y+r^2*s^3*x^2*y+r*s^3*x^3*y+s^3*x^4*
y+2*r^2*s^3*x*y+r^2*s^2*x^2*y+4*r^2*s^2*x*y^2+2*r*s^3*x^2*y+2*r*s^2*x^3*y+4*r*s^2*x^2*y^2+2*
s^3*x^3*y+3*s^2*x^4*y+4*s^2*x^3*y^2+8*r^2*s^2*x*y+2*r*s^3*x*y+6*r*s^2*x^2*y+4*r*s^2*x*y^2+2*
r*s*x^3*y+3*r*s*x^2*y^2+9*r*s*x*y^3+2*s^3*x^2*y+5*s^2*x^3*y+4*s^2*x^2*y^2+5*s*x^4*y+6*s*x^3*
y^2+9*s*x^2*y^3+7*r*s^2*x*y+5*r*s*x^2*y+17*r*s*x*y^2+4*s^2*x^2*y+7*s*x^3*y+10*s*x^2*y^2+5*
x^4*y+4*x^3*y^2+5*x^2*y^3+14*x*y^4+19*r*s*x*y+7*s*x^2*y+7*x^3*y+7*x^2*y^2+21*x*y^3+7*x^2*
y+20*x*y^2+13*x*y)*t^4+(r^4*s^4*x*y+r^3*s^4*x^2*y+r^2*s^4*x^3*y+r*s^4*x^4*y+s^4*x^5*y+3*r^3*
s^4*x*y+r^3*s^3*x^2*y+5*r^3*s^3*x*y^2+3*r^2*s^4*x^2*y+2*r^2*s^3*x^3*y+5*r^2*s^3*x^2*y^2+3*r*
s^4*x^3*y+3*r*s^3*x^4*y+5*r*s^3*x^3*y^2+3*s^4*x^4*y+4*s^3*x^5*y+5*s^3*x^4*y^2+13*r^3*s^3*x*
y+5*r^2*s^4*x*y+11*r^2*s^3*x^2*y+10*r^2*s^3*x*y^2+2*r^2*s^2*x^3*y+4*r^2*s^2*x^2*y^2+14*r^2*
s^2*x*y^3+5*r*s^4*x^2*y+10*r*s^3*x^3*y+10*r*s^3*x^2*y^2+5*r*s^2*x^4*y+8*r*s^2*x^3*y^2+14*r*
s^2*x^2*y^3+5*s^4*x^3*y+10*s^3*x^4*y+10*s^3*x^3*y^2+9*s^2*x^5*y+12*s^2*x^4*y^2+14*s^2*x^3*
y^3+23*r^2*s^3*x*y+9*r^2*s^2*x^2*y+38*r^2*s^2*x*y^2+5*r*s^4*x*y+17*r*s^3*x^2*y+10*r*s^3*x*
y^2+15*r*s^2*x^3*y+30*r*s^2*x^2*y^2+14*r*s^2*x*y^3+5*r*s*x^4*y+6*r*s*x^3*y^2+9*r*s*x^2*
```

```
y^3+28*r*s*x*y^4+5*s^4*x^2*y+14*s^3*x^3*y+10*s^3*x^2*y^2+19*s^2*x^4*y+25*s^2*x^3*y^2+14*s^2*
x^2*y^3+14*s*x^5*y+15*s*x^4*y^2+18*s*x^3*y^3+28*s*x^2*y^4+55*r^2*s^2*x*y+20*r*s^3*x*y+34*r*
s^2*x^2*y+33*r*s^2*x*y^2+13*r*s*x^3*y+20*r*s*x^2*y^2+67*r*s*x*y^3+12*s^3*x^2*y+24*s^2*x^3*
y+21*s^2*x^2*y^2+26*s*x^4*y+29*s*x^3*y^2+44*s*x^2*y^3+14*x^5*y+10*x^4*y^2+10*x^3*y^3+14*x^2*
y^4+42*x*y^5+44*r*s^2*x*y+26*r*s*x^2*y+97*r*s*x*y^2+19*s^2*x^2*y+31*s*x^3*y+43*s*x^2*y^2+26*
x^4*y+20*x^3*y^2+26*x^2*y^3+84*x*y^4+103*r*s*x*y+32*s*x^2*y+31*x^3*y+31*x^2*y^2+105*x*y^3+32*
x^2*y+100*x*y^2+68*x*y)*t^5+(r^5*s^5*x*y+r^4*s^5*x^2*y+r^3*s^5*x^3*y+r^2*s^5*x^4*y+r*s^5*x^5*
y+s^5*x^6*y+4*r^4*s^5*x*y+r^4*s^4*x^2*y+6*r^4*s^4*x*y^2+4*r^3*s^5*x^2*y+2*r^3*s^4*x^3*y+6*
r^3*s^4*x^2*y^2+4*r^2*s^5*x^3*y+3*r^2*s^4*x^4*y+6*r^2*s^4*x^3*y^2+4*r*s^5*x^4*y+4*r*s^4*x^5*
y+6*r*s^4*x^4*y^2+4*s^5*x^5*y+5*s^4*x^6*y+6*s^4*x^5*y^2+19*r^4*s^4*x*y+9*r^3*s^5*x*y+17*r^3*
s^4*x^2*y+18*r^3*s^4*x*y^2+2*r^3*s^3*x^3*y+5*r^3*s^3*x^2*y^2+20*r^3*s^3*x*y^3+9*r^2*s^5*x^2*
y+16*r^2*s^4*x^3*y+18*r^2*s^4*x^2*y^2+5*r^2*s^3*x^4*y+10*r^2*s^3*x^3*y^2+20*r^2*s^3*x^2*
y^3+9*r*s^5*x^3*y+16*r*s^4*x^4*y+18*r*s^4*x^3*y^2+9*r*s^3*x^5*y+15*r*s^3*x^4*y^2+20*r*s^3*
x^3*y^3+9*s^5*x^4*y+17*s^4*x^5*y+18*s^4*x^4*y^2+14*s^3*x^6*y+20*s^3*x^5*y^2+20*s^3*x^4*
y^3+51*r^3*s^4*x*y+14*r^3*s^3*x^2*y+70*r^3*s^3*x*y^2+14*r^2*s^5*x*y+41*r^2*s^4*x^2*y+30*r^2*
s^4*x*y^2+25*r^2*s^3*x^3*y+62*r^2*s^3*x^2*y^2+40*r^2*s^3*x*y^3+5*r^2*s^2*x^4*y+8*r^2*s^2*x^3*
y^2+14*r^2*s^2*x^2*y^3+48*r^2*s^2*x*y^4+14*r*s^5*x^2*y+35*r*s^4*x^3*y+30*r*s^4*x^2*y^2+33*r*
s^3*x^4*y+57*r*s^3*x^3*y^2+40*r*s^3*x^2*y^3+14*r*s^2*x^5*y+20*r*s^2*x^4*y^2+28*r*s^2*x^3*
y^3+48*r*s^2*x^2*y^4+14*s^5*x^3*y+33*s^4*x^4*y+30*s^4*x^3*y^2+41*s^3*x^5*y+56*s^3*x^4*y^2+40*
s^3*x^3*y^3+28*s^2*x^6*y+36*s^2*x^5*y^2+42*s^2*x^4*y^3+48*s^2*x^3*y^4+122*r^3*s^3*x*y+75*r^2*
s^4*x*y+91*r^2*s^3*x^2*y+124*r^2*s^3*x*y^2+21*r^2*s^2*x^3*y+42*r^2*s^2*x^2*y^2+157*r^2*s^2*x*
y^3+14*r*s^5*x*y+55*r*s^4*x^2*y+30*r*s^4*x*y^2+75*r*s^3*x^3*y+98*r*s^3*x^2*y^2+40*r*s^3*x*
y^3+46*r*s^2*x^4*y+72*r*s^2*x^3*y^2+128*r*s^2*x^2*y^3+48*r*s^2*x*y^4+14*r*s*x^5*y+15*r*s*x^4*
y^2+18*r*s*x^3*y^3+28*r*s*x^2*y^4+90*r*s*x*y^5+14*s^5*x^2*y+45*s^4*x^3*y+30*s^4*x^2*y^2+70*
s^3*x^4*y+82*s^3*x^3*y^2+40*s^3*x^2*y^3+72*s^2*x^5*y+91*s^2*x^4*y^2+108*s^2*x^3*y^3+48*s^2*
x^2*y^4+42*s*x^6*y+42*s*x^5*y^2+45*s*x^4*y^3+56*s*x^3*y^4+90*s*x^2*y^5+197*r^2*s^3*x*y+66*
r^2*s^2*x^2*y+293*r^2*s^2*x*y^2+65*r*s^4*x*y+120*r*s^3*x^2*y+108*r*s^3*x*y^2+97*r*s^2*x^3*
y+194*r*s^2*x^2*y^2+137*r*s^2*x*y^3+41*r*s*x^4*y+49*r*s*x^3*y^2+76*r*s*x^2*y^3+260*r*s*x*
y^4+40*s^4*x^2*y+85*s^3*x^3*y+72*s^3*x^2*y^2+110*s^2*x^4*y+138*s^2*x^3*y^2+94*s^2*x^2*y^3+98*
s*x^5*y+98*s*x^4*y^2+115*s*x^3*y^3+184*s*x^2*y^4+42*x^6*y+28*x^5*y^2+25*x^4*y^3+28*x^3*
y^4+42*x^2*y^5+132*x*y^6+374*r^2*s^2*x*y+157*r*s^3*x*y+201*r*s^2*x^2*y+231*r*s^2*x*y^2+78*r*
s*x^3*y+123*r*s*x^2*y^2+448*r*s*x*y^3+71*s^3*x^2*y+129*s^2*x^3*y+113*s^2*x^2*y^2+141*s*x^4*
y+149*s*x^3*y^2+229*s*x^2*y^3+98*x^5*y+67*x^4*y^2+67*x^3*y^3+98*x^2*y^4+330*x*y^5+280*r*s^2*
x*y+147*r*s*x^2*y+589*r*s*x*y^2+103*s^2*x^2*y+161*s*x^3*y+219*s*x^2*y^2+141*x^4*y+105*x^3*
y^2+141*x^2*y^3+504*x*y^4+613*r*s*x*y+171*s*x^2*y+161*x^3*y+161*x^2*y^2+595*x*y^3+171*x^2*
y+570*x*y^2+399*x*y)*t^6:
```

```
> # We check against exhaustive generation:
expand(series(devJ-vJexh,t=0,6));
```

$$O(t^6) \tag{3.1.2}$$

```
> # Now we forget about r and s and we keep only one variable w marking the quantity (upper height)-3(lower height)
  # This amounts to the substitution s=w, r=1/w^3
  # We obtain equation (33)
  eqq3:=subs(s=w,r=1/w^3,J=M,j1=m1,j11=m11,eqq2);
```

$$eqq3 := -M + \frac{(Fy - y)\, x}{y} + \frac{t\, x\, (m1 - m11)\, Fy}{w^2\, (y-1)} + \frac{w\, t\, x^2\, (M - m1)\, Fy}{y\, (x-1)} + \frac{x\, t\, (Fx - f)\, M}{x - 1} \quad (3.1.3$$

```
> # Our equation has now two catalytic variables but it is still linear, let us look at the kernel:
  collect(eqq3,M):
  # bivariate kernel
  ker2:=coeff(%,M,1);
  # and remainder term
  rem2:=coeff(%%,M,0);

  # Since everything with one variable is known from previous works, we can express the kernel in the new variables
  # Here we introduce a new variable v which is to y what u is to x.
  subs(F=Fx,u=u,{eq1,eq2,eq3,eq4});
  subs(F=Fy,u=v,x=y,{eq1,eq2,eq3,eq4});
  sys2:=% union %%;
  eliminate(sys2 union {KK=ker2},{Fx,Fy,x,y,f,t})[2,1]:

  # Here is the kernel (more precisely its numerator) in the new variables!
  factor(numer(factor(solve(%,KK)))):
  ker2N:=map(x->collect(x,u),%);
```

$$ker2 := -1 + \frac{w\, t\, x^2\, Fy}{y\, (x-1)} + \frac{x\, t\, (Fx - f)}{x - 1}$$

$$rem2 := \frac{(Fy - y)\, x}{y} + \frac{t\, x\, (m1 - m11)\, Fy}{w^2\, (y-1)} - \frac{w\, t\, x^2\, m1\, Fy}{y\, (x-1)}$$

$$\left\{ -Fx + \frac{(1+u)\left(-u\, z^2 - 2\, z + 1\right)}{(u\, z + 1)\, (1 - z)^3},\ -f + \frac{1 - 2\, z}{(1 - z)^3},\ -t + z\, (1 - z)^3,\ -x + \frac{1 + u}{(u\, z + 1)^2} \right\}$$

$$\left\{-Fy+\frac{(1+v)\left(-v z^2-2z+1\right)}{(vz+1)(1-z)^3},\ -f+\frac{1-2z}{(1-z)^3},\ -t+z(1-z)^3,\ -y+\frac{1+v}{(vz+1)^2}\right\}$$

$$sys2 := \left\{-Fx+\frac{(1+u)\left(-u z^2-2z+1\right)}{(uz+1)(1-z)^3},\ -Fy+\frac{(1+v)\left(-v z^2-2z+1\right)}{(vz+1)(1-z)^3},\ -f+\frac{1-2z}{(1-z)^3},\ -t+z(1-z)^3,\ -x+\frac{1+u}{(uz+1)^2},\ -y+\frac{1+v}{(vz+1)^2}\right\}$$

$$ker2N := u^2 v^2 w z^4+2 u v^2 w z^4+3 u^2 v w z^3+v^2 w z^4-u^2 v w z^2-u^2 z^4+6 u v w z^3+2 u^2 w z^2+3 u^2 z^3-2 u v w z^2+3 v w z^3 -u^2 w z-3 u^2 z^2+4 u w z^2-u z^3-v w z^2+u^2 z-2 u w z+3 u z^2+2 w z^2-3 u z-w z+u \quad (3.1.4)$$

```
> # Now let us look at the two roots root U=u(z,v,w) of the kernel
  solve(ker2N,u):
  valU:=[%][1];
  otherU:=[%%][2]:
  # One of the roots is a non-singular power series in z (call it U_0)
  devU:=simplify(series(valU,z=0,5));
  # Note that the other root if NOT a power series
  simplify(series(otherU,z=0,3));
```

$$valU := \frac{1}{2}\,\frac{1}{z\left(v^2 w z^3+3 v w z^2-v w z-z^3+2 w z+3 z^2-w-3 z+1\right)}\Big(-2 v^2 w z^4-6 v w z^3+2 v w z^2-4 w z^2+z^3+2 w z -3 z^2+3 z-1 + \big(4 v^2 w z^8-16 v^2 w z^7+24 v^2 w z^6+12 v w z^7-16 v^2 w z^5-52 v w z^6+4 v^2 w z^4+88 v w z^5+8 w z^6-72 v w z^4 -36 w z^5+z^6+28 v w z^3+64 w z^4-6 z^5-4 v w z^2-56 w z^3+15 z^4+24 w z^2-20 z^3-4 w z+15 z^2-6 z+1\big)^{1/2}\Big)$$

$$devU := w z+(2w+1+v)\, w z^2+(5w+4v+3)\, w^2 z^3+O\left(z^4\right)$$

$$\frac{\frac{1}{w-1}}{z}-\frac{w\,(2w-1+v)}{(w-1)^2}+\frac{w\left(v^2 w-w^3+4 v w+3 w^2-2 v-1\right)}{(w-1)^3}\, z+O\left(z^2\right) \quad (3.1.5)$$

```
> # Now we will look at the cancellation of the remainder term
  # We first express this remainder in the new variables
  # The obtained equation has 99 terms.
  factor((subs(t=solve(eq1,t),f=solve(eq3,f),rem2))):
  subs(y=subs(u=v,solve(eq2,x)),%):
```

```
simplify(subs(Fx=solve(eq4,F),Fy=subs(u=v,solve(eq4,F)),%)):
rem2N:=factor(subs(x=solve(eq2,x),%)*v*(v*z+1)*(-1+z)^3*(u*z^2+2*z-1)*u*(u*z+1)^4/(1+u*z)^2*
w^2/(1+u)/z/(1+v*z));

nops(%);
```

$$
\begin{aligned}
rem2N := &-m1\,u\,v^3 w^3 z^6 + 3\,m1\,u\,v^3 w^3 z^5 - m1\,v^3 w^3 z^6 - 3\,m1\,u\,v^3 w^3 z^4 - 3\,m1\,u\,v^2 w^3 z^5 + 3\,m1\,v^3 w^3 z^5 + m1\,u^2 v^2 z^6 + m1\,u\,v^3 w^3 z^3 \\
&+ 10\,m1\,u\,v^2 w^3 z^4 - 3\,m1\,v^3 w^3 z^4 - 3\,m1\,v^2 w^3 z^5 - m11\,u^2 v^2 z^6 + u^2 v^3 w^2 z^4 - 3\,m1\,u^2 v^2 z^5 + m1\,u^2 v\,z^6 - 12\,m1\,u\,v^2 w^3 z^3 \\
&- 2\,m1\,u\,v\,w^3 z^4 + m1\,v^3 w^3 z^3 + 10\,m1\,v^2 w^3 z^4 + 3\,m11\,u^2 v^2 z^5 - m11\,u^2 v\,z^6 + 3\,m1\,u^2 v^2 z^4 - 2\,m1\,u^2 v\,z^5 + 6\,m1\,u\,v^2 w^3 z^2 \\
&+ 2\,m1\,u\,v^2 z^5 + 7\,m1\,u\,v\,w^3 z^3 - 12\,m1\,v^2 w^3 z^3 - 2\,m1\,v\,w^3 z^4 - 3\,m11\,u^2 v^2 z^4 + 2\,m11\,u^2 v\,z^5 - 2\,m11\,u\,v^2 z^5 + 3\,u^2 v^2 w^2 z^3 \\
&- u^2 v\,w^2 z^4 + 2\,u\,v^3 w^2 z^3 - m1\,u^2 v^2 z^3 + m1\,u^2 z^5 - m1\,u\,v^2 w^3 z - 7\,m1\,u\,v^2 z^4 - 9\,m1\,u\,v\,w^3 z^2 + 2\,m1\,u\,v\,z^5 + 6\,m1\,v^2 w^3 z^2 \\
&+ 7\,m1\,v\,w^3 z^3 + m11\,u^2 v^2 z^3 - m11\,u^2 z^5 + 7\,m11\,u\,v^2 z^4 - 2\,m11\,u\,v\,z^5 - u^2 v^2 w^2 z^2 + 3\,u^2 v\,w^2 z^3 - u\,v^3 w^2 z^2 + 2\,m1\,u^2 v\,z^3 \\
&- 3\,m1\,u^2 z^4 + 9\,m1\,u\,v^2 z^3 + 5\,m1\,u\,v\,w^3 z - 5\,m1\,u\,v\,z^4 - m1\,v^2 w^3 z - 9\,m1\,v\,w^3 z^2 - 2\,m11\,u^2 v\,z^3 + 3\,m11\,u^2 z^4 - 9\,m11\,u\,v^2 z^3 \\
&+ 5\,m11\,u\,v\,z^4 - u^2 v\,w^2 z^2 + 6\,u\,v^2 w^2 z^2 - 2\,u\,v\,w^2 z^3 - m1\,u^2 v\,z^2 + 3\,m1\,u^2 z^3 - 5\,m1\,u\,v^2 z^2 - m1\,u\,v\,w^3 + 2\,m1\,u\,v\,z^3 + 2\,m1\,u\,z^4 \\
&+ 5\,m1\,v\,w^3 z + m11\,u^2 v\,z^2 - 3\,m11\,u^2 z^3 + 5\,m11\,u\,v^2 z^2 - 2\,m11\,u\,v\,z^3 - 2\,m11\,u\,z^4 - 5\,u\,v^2 w^2 z + 7\,u\,v\,w^2 z^2 - m1\,u^2 z^2 \\
&+ m1\,u\,v^2 z + 4\,m1\,u\,v\,z^2 - 7\,m1\,u\,z^3 - m1\,v\,w^3 + m11\,u^2 z^2 - m11\,u\,v^2 z - 4\,m11\,u\,v\,z^2 + 7\,m11\,u\,z^3 + u\,v^2 w^2 - 5\,u\,v\,w^2 z \\
&- 4\,m1\,u\,v\,z + 9\,m1\,u\,z^2 + 4\,m11\,u\,v\,z - 9\,m11\,u\,z^2 + u\,v\,w^2 + m1\,u\,v - 5\,m1\,u\,z - m11\,u\,v + 5\,m11\,u\,z + m1\,u - m11\,u
\end{aligned}
$$

99 (3.1.6

```
> # We can eliminate U0 between the kernel and the remainder, and (up to trivial factors) we
end up with an algebraic equation in which u and \tilde{M}(u,v)
# do not appear anymore. This is Eq (37) in the paper:

eqB:=factor(eliminate({rem2N, ker2N},u)[2,1]/(w*(-1+z)*(v*z^2+2*z-1)*(v*z+1))):

# The equation has 854 terms
nops(%);

# The equation is not linear but quadratic!
degree(%%,m1);
```

854

2 (3.1.7

```
> # At this stage we have obtained an equation, eqB, in which the variables M and u (or x) is
```

```
no longer present
# This equation involves variables m1=M(1,y), m11=M(1,1), v=v(y), and w,z
# This equation is quadratic in m's so we can apply (a variant of) the quadratic method
# We will try to cancel both the equation and its derivative with respect to m1
# Explicitly, we have these two equations (we remove trivial factors from the second one):
E1:=eqB:
E2:=factor(diff(eqB,m1)/(1-z)^3):
```

```
> # (short interplay: we will soon need to look at expansions, so let us record the expansions
of M1 and m11 obtained from the functional equation)

subs(t=solve(eq1,t),y=y+subs(x=y,u=v,eq2),subs(x=1,convert(devJ,polynom))):
devJ1_z:=simplify(subs(s=w,r=1/w^3,series(%,z=0,4)));
subs(t=solve(eq1,t),subs(x=1,y=1,convert(devJ,polynom))):
devJ11_z:=simplify(subs(s=w,r=1/w^3,series(%,z=0,5)));
subs(m1=devJ1_z,m11=devJ11_z,E2):
factor(mtaylor(%,[z,v],4));
```

$$devJ1\_z := (1+v)\,z + \frac{v\,w^3 + v\,w^2 + w^3 + w^2 + v + 1}{w^2}\,z^2 + \frac{v^2 w^5 + 2\,v\,w^6 - v^2 w^4 + 2\,v\,w^5 + 2\,w^6 - 3\,v\,w^4 + w^5 + v^2 w^2 + 2\,v\,w^3 - 2\,w^4 + 3\,v\,w^2 + 2\,w^3 + 2\,w^2 + v + 1}{w^4}\,z^3 + \mathrm{O}(z^4)$$

$$devJ11\_z := z + \frac{w^3 + w^2 + 1}{w^2}\,z^2 + \frac{2\,w^6 + w^5 - 2\,w^4 + 2\,w^3 + 2\,w^2 + 1}{w^4}\,z^3 + \frac{5\,w^9 + 2\,w^8 - 4\,w^7 - 4\,w^6 + 5\,w^5 - 2\,w^4 + 3\,w^3 + 4\,w^2 + 1}{w^6}\,z^4 + \mathrm{O}(z^5)$$

$$-v\,(w-1)\,\left(-2\,v\,w^5 z + v^2 w^4 - 6\,v\,w^4 z + v\,w^4 + w^3 z^2 - v\,w^2 z + 5\,w^2 z^2 - w^2 z - z^2\right) \qquad \text{(3.1.8}$$

```
> ### We prove that there is a formal power series V_0(z) that cancels E2.
### To see this we only have to check the first terms of the expansion (which we know from
the expansions above)
# We look at the coefficients of m1^2 and m1^1 in the main equation eqB
aa:=subs(v=v,coeff(E1, m1, 2)):
bb:=subs(v=v,coeff(E1, m1, 1)):
```

```
# and we look at all the sub-cubic monomials in variables z,v
map(factor,mtaylor(aa,[z,v],4));
map(factor,mtaylor(bb,[z,v],4));

# substituting M(1,y) and M(1,1) by their expansion, we see that the equation E2 near z=0 is
of the following form, up to at-least-cubic terms:
2*aa*m1+bb:
subs(m1= subs(s=w,r=1/w^3,(1+v)*z+(r*s*v+r*s+s*v+s+v+1)*z^2+O(z^3)) ,m11=subs(s=w,r=1/w^3,z+
(r*s+s+1)*z^2+O(z^3)),%):
map(simplify,mtaylor(%,[z,v],4));
factor(%/v);
algcurves[puiseux](RootOf(%,v),z=0,2);
# It follows by a Newton-Puiseux argument that there is a root starting with O(z) as
justified in the paper
```

$$-w^2 (w-1) v + (w-1) z + w^2 (w^2+w+1) (w-1)^2 v^2 + (w-1) (9 w^2+2) v z - 8 (w-1) z^2 - (w-1) (10 w^5 - w^4 - w^3 - 8 w^2 - 1) z v^2 - (w-1) (37 w^2 + w + 14) v z^2 + 27 (w-1) z^3$$

$$m11\, w^2 (w-1) v - 2\, m11 (w-1) z + w^2 (w-1) (-w^2 + m11) v^2 - (w-1) (9\, m11\, w^2 - 2 w^2 + 4\, m11) v z + 16\, m11 (w-1) z^2 - w^4 (w-1) v^3 - (-9 w^5 + 7\, m11\, w^3 + 9 w^4 - 8\, m11\, w^2 - 4 w^3 + 2\, m11\, w + 4 w^2 - 2\, m11) z v^2 + (37\, m11\, w^3 - 36\, m11\, w^2 - 16 w^3 + 26\, m11\, w + 16 w^2 - 28\, m11) v z^2 - 54\, m11 (w-1) z^3$$

$$-w^4 (w-1) v^2 + w^2 (w-1) v z - (w^4 + 7 w^3 - 8 w^2 - w + 1) v z^2 + w^2 (2 w^4 + 7 w^3 - 9 w^2 + w - 1) z v^2 - w^4 (w-1) v^3$$

$$-(w-1) (-2 v w^5 z + v^2 w^4 - 9 v w^4 z + v w^4 + w^3 z^2 - v w^2 z + 8 w^2 z^2 - w^2 z - z^2)$$

$$\left\{ \frac{z}{w^2}, -1 + (2 w + 9) z \right\} \qquad (3.1.9$$

```
> # Check of the justification in the notation of the paper:
-w^4*(w-1)*v^2+w^2*(w-1)*v*z-(w^4+7*w^3-8*w^2-w+1)*v*z^2+w^2*(2*w^4+7*w^3-9*w^2+w-1)*v^2*z-
w^4*(w-1)*v^3;
a[2,0]*v^2+a[1,1]*v*z+a[1,2]*v*z^2+a[2,1]*v^2*z+a[3,0]*v^3 +O4;
subs(O4=a*z^4+b*z^3*v+c*z^2*v^2+d*z*v^3+e*v^4,%);
algcurves[puiseux](RootOf(%,v),z=0,2);
```

$$-w^4 (w-1) v^2 + w^2 (w-1) v z - (w^4 + 7 w^3 - 8 w^2 - w + 1) v z^2 + w^2 (2 w^4 + 7 w^3 - 9 w^2 + w - 1) z v^2 - w^4 (w-1) v^3$$

$$v^3 a_{3,0} + v^2 z a_{2,1} + v z^2 a_{1,2} + v^2 a_{2,0} + v z a_{1,1} + O4$$

$$a z^4 + b v z^3 + c v^2 z^2 + d v^3 z + e v^4 + v^3 a_{3,0} + v^2 z a_{2,1} + v z^2 a_{1,2} + v^2 a_{2,0} + v z a_{1,1}$$

$$\left\{0, -\frac{z a_{1,1}}{a_{2,0}}, \frac{1}{4 e a_{2,0}^2 - a_{2,0} a_{3,0}^2} \left(\left(-RootOf\left(\_Z^2 e + \_Z a_{3,0} + a_{2,0}\right) d\, a_{2,0} a_{3,0} - RootOf\left(\_Z^2 e + \_Z a_{3,0} + a_{2,0}\right) e\, a_{1,1} a_{3,0} + 2\, RootOf\left(\_Z^2 e + \_Z a_{3,0} + a_{2,0}\right) e\, a_{2,0} a_{2,1} - 2 d\, a_{2,0}^2 + 2 e\, a_{1,1} a_{2,0} - a_{1,1} a_{3,0}^2 + a_{2,0} a_{2,1} a_{3,0} \right) z \right) + RootOf\left(\_Z^2 e + \_Z a_{3,0} + a_{2,0}\right) \right\} \quad (3.1.10$$

```
> # substituting v=V0 we cancel both E1 and E2.
  # We can then eliminate m1=M(1,y) between E1 and E2, which here just amounts to taking the
  discriminant:
  factor(discrim(E1,m1));
```

$$(-1+z)^6 (1+v)^2 (4 v^2 w z^4 + 12 v w z^3 - 4 v w z^2 + 8 w z^2 - 4 w z + z^2 - 2 z + 1) (v^2 w z^3 + 3 v w z^2 - v w z + 2 w z + z^2 - w - 2 z + 1)^2 (m11\, v z^4 - v^2 w^2 z^2 - 3\, m11\, v z^3 + v w^2 z^2 + 3\, m11\, v z^2 + m11\, z^3 - 3 v w^2 z - m11\, v z - 3\, m11\, z^2 + v w^2 + 3\, m11\, z - m11)^2 v^2 w^4 \quad (3.1.11$$

```
> # Disregarding trivial factors, we have three irreducible factors P1,P2,P3:
  discrimE:=factor(discrim(E1,m1)/(-1+z)^6/(1+v)^2/w^4/v^2);
  P1:=op(1,discrimE);
  P2:=op(1,op(2,discrimE));
  P3:=op(1,op(3,discrimE));
```

$$discrimE := (4 v^2 w z^4 + 12 v w z^3 - 4 v w z^2 + 8 w z^2 - 4 w z + z^2 - 2 z + 1) (v^2 w z^3 + 3 v w z^2 - v w z + 2 w z + z^2 - w - 2 z + 1)^2 (m11\, v z^4 - v^2 w^2 z^2 - 3\, m11\, v z^3 + v w^2 z^2 + 3\, m11\, v z^2 + m11\, z^3 - 3 v w^2 z - m11\, v z - 3\, m11\, z^2 + v w^2 + 3\, m11\, z - m11)^2$$

$$P1 := 4 v^2 w z^4 + 12 v w z^3 - 4 v w z^2 + 8 w z^2 - 4 w z + z^2 - 2 z + 1$$

$$P2 := v^2 w z^3 + 3 v w z^2 - v w z + 2 w z + z^2 - w - 2 z + 1$$

$$P3 := m11\, v z^4 - v^2 w^2 z^2 - 3\, m11\, v z^3 + v w^2 z^2 + 3\, m11\, v z^2 + m11\, z^3 - 3 v w^2 z - m11\, v z - 3\, m11\, z^2 + v w^2 + 3\, m11\, z - m11 \quad (3.1.12$$

```
> # P1 and P2 do not involve m11, let us look at their v-roots:
  map(simplify,algcurves[puiseux](RootOf(P1,v),z=0,0));
  map(simplify,algcurves[puiseux](RootOf(P2,v),z=0,1));
  expand(%);

  # None of them is a formal power series so V_0 is not one of these roots!
```

```
# This proves that P3=0 when v=V_0.
```

$$\left\{ \frac{RootOf(4\,\_Z^2 w - 4\,\_Z w + 1)}{z^2} \right\}$$

$$\left\{ -\frac{2 w^2 z - w z^2 - w^2 + w z + z^2}{w^2 z^2},\ -\frac{(w-1)(z+w)}{z w^2} \right\}$$

$$\left\{ -\frac{1}{z} - \frac{1}{w} + \frac{1}{w z} + \frac{1}{w^2},\ -\frac{2}{z} + \frac{1}{w} + \frac{1}{z^2} - \frac{1}{w z} - \frac{1}{w^2} \right\} \quad (3.1.13$$

```
> # We now solve P3 for m11
M11_rat:=solve(P3, m11);
# This gives us a rational expression of M(1,1) in terms of the series V_0 above.
# Note that at this stage V_0 is still unknown (we only know it exists)
# This is the rational function R(z,V) given in the paper in Eq. (39) in the paper.
```

$$M11\_rat := \frac{(v z^2 - z^2 + 3 z - 1)\, v w^2}{v z^4 - 3 v z^3 + 3 v z^2 + z^3 - v z - 3 z^2 + 3 z - 1} \quad (3.1.14$$

```
> # We now substitute this expression of m11 into E1,E2:
factor(numer(subs(m11=M11_rat,E1)));
factor(numer(subs(m11=M11_rat,E2)));

# Up to trivial factors z,1+z, 1-w, 1+v, these two equations each have the same nonstrivial polynomial factor!
# We call this factor S, it gives us a polynomial equation satisfied by v=V_0. This is the polynomial S(V,z,w) given in Eq. (40) in the paper.
# We have thus proved Theorem 4.2

S_factor:=op(6,%);
```

$$m1^2 (-1+z)^{12} (w-1) (v z+1)^3 \big(v^5 w^5 z^5 + v^5 w^4 z^5 + v^5 w^3 z^5 + 5 v^4 w^5 z^4 + v^5 w^2 z^5 - 2 v^4 w^5 z^3 + 5 v^4 w^4 z^4 - v^3 w^4 z^5 + v^5 w z^5$$
$$- 2 v^4 w^4 z^3 + 5 v^4 w^3 z^4 + 2 v^4 w^2 z^5 + 8 v^3 w^5 z^3 + 3 v^3 w^4 z^4 - v^3 w^3 z^5 - 2 v^4 w^3 z^3 + 4 v^4 w^2 z^4 + 2 v^4 w z^5 - 6 v^3 w^5 z^2 + 5 v^3 w^4 z^3$$
$$+ 3 v^3 w^3 z^4 + v^3 w^2 z^5 - 2 v^2 w^4 z^4 - v^4 w^2 z^3 + 4 v^4 w z^4 + v^3 w^5 z - 5 v^3 w^4 z^2 + 5 v^3 w^3 z^3 + 8 v^3 w^2 z^4 + v^3 w z^5 + 4 v^2 w^5 z^2$$

$$+7 v^2 w^4 z^3 - 2 v^2 w^3 z^4 - v^4 w z^3 + v^3 w^4 z - 5 v^3 w^3 z^2 + v^3 w^2 z^3 + 8 v^3 w z^4 - 4 v^2 w^5 z - 5 v^2 w^4 z^2 + 7 v^2 w^3 z^3 + 4 v^2 w^2 z^4$$
$$+ v^3 w^3 z - v^3 w^2 z^2 + v^3 w z^3 + v^2 w^5 + v^2 w^4 z - 5 v^2 w^3 z^2 + 6 v^2 w^2 z^3 + 4 v^2 w z^4 - v^3 w z^2 - 2 v^3 z^3 + v^2 w^3 z - 5 v^2 w^2 z^2$$
$$+ 5 v^2 w z^3 + 4 v w^2 z^3 + v^3 z^2 + 3 v^2 w^2 z - 2 v^2 w z^2 - 4 v^2 z^3 - 4 v w^2 z^2 + 3 v w z^3 - v^2 w^2 + 3 v w^2 z - v w z^2 - 2 v z^3 + v^2 z$$
$$- v w^2 - 3 v z^2 + 2 v z - 2 z^2 + z\big)$$
$$-2\, m1\, (-1+z)^6 (w-1)(v z+1)^2 \big(v^5 w^5 z^5 + v^5 w^4 z^5 + v^5 w^3 z^5 + 5 v^4 w^5 z^4 + v^5 w^2 z^5 - 2 v^4 w^5 z^3 + 5 v^4 w^4 z^4 - v^3 w^4 z^5 + v^5 w z^5$$
$$- 2 v^4 w^4 z^3 + 5 v^4 w^3 z^4 + 2 v^4 w^2 z^5 + 8 v^3 w^5 z^3 + 3 v^3 w^4 z^4 - v^3 w^3 z^5 - 2 v^4 w^3 z^3 + 4 v^4 w^2 z^4 + 2 v^4 w z^5 - 6 v^3 w^5 z^2 + 5 v^3 w^4 z^3$$
$$+ 3 v^3 w^3 z^4 + v^3 w^2 z^5 - 2 v^2 w^4 z^4 - v^4 w^2 z^3 + 4 v^4 w z^4 + v^3 w^5 z - 5 v^3 w^4 z^2 + 5 v^3 w^3 z^3 + 8 v^3 w^2 z^4 + v^3 w z^5 + 4 v^2 w^5 z^2$$
$$+7 v^2 w^4 z^3 - 2 v^2 w^3 z^4 - v^4 w z^3 + v^3 w^4 z - 5 v^3 w^3 z^2 + v^3 w^2 z^3 + 8 v^3 w z^4 - 4 v^2 w^5 z - 5 v^2 w^4 z^2 + 7 v^2 w^3 z^3 + 4 v^2 w^2 z^4$$
$$+ v^3 w^3 z - v^3 w^2 z^2 + v^3 w z^3 + v^2 w^5 + v^2 w^4 z - 5 v^2 w^3 z^2 + 6 v^2 w^2 z^3 + 4 v^2 w z^4 - v^3 w z^2 - 2 v^3 z^3 + v^2 w^3 z - 5 v^2 w^2 z^2$$
$$+ 5 v^2 w z^3 + 4 v w^2 z^3 + v^3 z^2 + 3 v^2 w^2 z - 2 v^2 w z^2 - 4 v^2 z^3 - 4 v w^2 z^2 + 3 v w z^3 - v^2 w^2 + 3 v w^2 z - v w z^2 - 2 v z^3 + v^2 z$$
$$- v w^2 - 3 v z^2 + 2 v z - 2 z^2 + z\big)$$

$$S\_factor := v^5 w^5 z^5 + v^5 w^4 z^5 + v^5 w^3 z^5 + 5 v^4 w^5 z^4 + v^5 w^2 z^5 - 2 v^4 w^5 z^3 + 5 v^4 w^4 z^4 - v^3 w^4 z^5 + v^5 w z^5 - 2 v^4 w^4 z^3 + 5 v^4 w^3 z^4$$
$$+ 2 v^4 w^2 z^5 + 8 v^3 w^5 z^3 + 3 v^3 w^4 z^4 - v^3 w^3 z^5 - 2 v^4 w^3 z^3 + 4 v^4 w^2 z^4 + 2 v^4 w z^5 - 6 v^3 w^5 z^2 + 5 v^3 w^4 z^3 + 3 v^3 w^3 z^4 + v^3 w^2 z^5$$
$$- 2 v^2 w^4 z^4 - v^4 w^2 z^3 + 4 v^4 w z^4 + v^3 w^5 z - 5 v^3 w^4 z^2 + 5 v^3 w^3 z^3 + 8 v^3 w^2 z^4 + v^3 w z^5 + 4 v^2 w^5 z^2 + 7 v^2 w^4 z^3 - 2 v^2 w^3 z^4$$
$$- v^4 w z^3 + v^3 w^4 z - 5 v^3 w^3 z^2 + v^3 w^2 z^3 + 8 v^3 w z^4 - 4 v^2 w^5 z - 5 v^2 w^4 z^2 + 7 v^2 w^3 z^3 + 4 v^2 w^2 z^4 + v^3 w^3 z - v^3 w^2 z^2 + v^3 w z^3$$
$$+ v^2 w^5 + v^2 w^4 z - 5 v^2 w^3 z^2 + 6 v^2 w^2 z^3 + 4 v^2 w z^4 - v^3 w z^2 - 2 v^3 z^3 + v^2 w^3 z - 5 v^2 w^2 z^2 + 5 v^2 w z^3 + 4 v w^2 z^3 + v^3 z^2$$
$$+ 3 v^2 w^2 z - 2 v^2 w z^2 - 4 v^2 z^3 - 4 v w^2 z^2 + 3 v w z^3 - v^2 w^2 + 3 v w^2 z - v w z^2 - 2 v z^3 + v^2 z - v w^2 - 3 v z^2 + 2 v z - 2 z^2 + z \quad \text{(3.1.15)}$$

```
> # As mentionned after the theorem, we have two equations, we can eliminate V_0 and we get an explicit algebraic equation for the series M(1,1) .
final_eq_m11:=eliminate({m11=M11_rat, S_factor},{v})[2,1]:
# It has 601 terms and degree 5
nops(%);
degree(%%,m11);
```

$$601$$
$$5 \quad \text{(3.1.16)}$$

```
> # We can check its expansion:, it works!
series(RootOf(final_eq_m11,m11),z=0,5);
simplify(series(devJ11_z-%,z=0)):
subs(s=w,r=1/w^3,%);
```

$$z+\frac{w^3+w^2+1}{w^2}z^2+\frac{(w+1)\left(2w^5-w^4-w^3+3w^2-w+1\right)}{w^4}z^3$$
$$+\frac{5w^9+2w^8-4w^7-4w^6+5w^5-2w^4+3w^3+4w^2+1}{w^6}z^4+\mathrm{O}\left(z^5\right)$$
$$\mathrm{O}\left(z^5\right) \quad (3.1.17$$

Section 5: asymptotic checks

Section 5.1: Proof of Thm 1.1

```
> #### PRELIMINARIES
  # First, we check that the main singularity of the classical series is at rho:=27/256
  factor(discrim(eq1,z));
  # This points corresponds to z(t)=1/4
  factor(subs(t=27/256,eq1));
```

$$t^2\,(256\,t-27)$$
$$-\frac{1}{256}\left(16z^2-40z+27\right)(-1+4z)^2 \quad (4.1.1$$

```
> #We start by rewriting our equation for h=H(1)=H(t,1,s) in terms of the variable delta:
  delta^2=1-4*z;
  valz_delta:=solve(%,z);
  eqh_deltas:=factor(numer(subs(z=valz_delta,final_eq_hzs)));
```

$$\delta^2=1-4z$$
$$valz\_delta:=-\frac{1}{4}\delta^2+\frac{1}{4}$$
$$eqh\_deltas:=\delta^{20}h^3s^3+26\delta^{18}h^3s^3+297\delta^{16}h^3s^3+64\delta^{16}h^2s^2+1944\delta^{14}h^3s^3+1152\delta^{14}h^2s^2+7938\delta^{12}h^3s^3+8832\delta^{12}h^2s^2 \quad (4.1.2$$
$$+20412\delta^{10}h^3s^3-256\delta^{12}hs^2+512\delta^{12}hs+38016\delta^{10}h^2s^2+30618\delta^8h^3s^3-256\delta^{12}h-4608\delta^{10}hs^2+1024\delta^{10}hs$$

$$+ 103680\,\delta^8 h^2 s^2 + 17496\,\delta^6 h^3 s^3 - 4608\,\delta^{10} h - 34560\,\delta^8 h s^2 - 16896\,\delta^8 h s + 196992\,\delta^6 h^2 s^2 - 19683\,\delta^4 h^3 s^3 - 34560\,\delta^8 h - 138240\,\delta^6 h s^2 - 67584\,\delta^6 h s + 279936\,\delta^4 h^2 s^2 - 39366\,\delta^2 h^3 s^3 - 138240\,\delta^6 h - 311040\,\delta^4 h s^2 - 41472\,\delta^4 h s + 279936\,\delta^2 h^2 s^2 - 19683\,h^3 s^3 - 311040\,\delta^4 h - 373248\,\delta^2 h s^2 + 262144\,\delta^4 + 82944\,\delta^2 h s + 139968\,h^2 s^2 - 373248\,\delta^2 h - 186624\,h s^2 + 524288\,\delta^2 + 41472\,h s - 186624\,h + 262144$$

```
> # For further use it will be useful to express the quantity 1-256/27t (called "increment"
  below) in terms of delta:
  increment:=factor(solve(eliminate({1-256/27*t=dt,eq1, delta^2=1-4*z},{t,z})[2,1],dt));

  # ... and to record the equation linking delta to t:
  eq_delta:=eliminate({delta^2=1-4*z,eq1},z)[2,1];
```

$$increment := \frac{1}{27}\,\delta^4\left(\delta^4 + 8\,\delta^2 + 18\right)$$

$$eq\_delta := \delta^8 + 8\,\delta^6 + 18\,\delta^4 + 256\,t - 27 \qquad \text{(4.1.3}$$

```
> #### Recurrence formula
  # Since the function H(s=1+r) is algebraic, it is D-finite, i.e. its derivatives at s=1
  satisfy a linear recurrence relation
  # which can be computed automatically by the (great) package gfun
  eqh_deltas:
  subs(s=r+1,%):
  gfun[algeqtodiffeq](%,h(r))[1]:
  # here c(k) is 1/k! times the k-th derivative, i.e. c(k)= 1/k! ((d/ds)^k h)(s=1) expressed in
  the variable delta
  map(factor,gfun[diffeqtorec](%,h(r),c(k))[1]):

  subs(seq(c(k-i)=1/(k-i)!*f(k-i),i=0..6),subs(k=k-6,%)):

  # We thus obtain the wanted recurrence to compute the derivatites f(k) (denoted by LaTeX "f^{
  (k)}" in the paper)
  # f(k)=RHS
  # with
  RHS:=map(factor,collect(expand(solve(%,f(k))),f));
```

$$RHS := -\frac{1}{4096}\,\frac{(2k-11)\,(k-2)\,(k-3)\,(k-4)\,(k-5)^2\,(\delta-1)\,(\delta+1)\,(\delta^4+3)\,(\delta^2+3)^4\,f(k-6)}{\delta^{10}} - \frac{1}{4096}\,\frac{1}{\delta^{10}}\big((k \qquad \text{(4.1.4}$$

$$-2)\,(k-3)\,(k-4)\,\left(\delta^2+3\right)\,\left(4\,\delta^{12}\,k^2-41\,\delta^{12}\,k+103\,\delta^{12}+224\,\delta^{10}\,k^2-1960\,\delta^{10}\,k+4280\,\delta^{10}+404\,\delta^8\,k^2-3261\,\delta^8\,k+6499\,\delta^8+672\,\delta^6\,k^2-5880\,\delta^6\,k+12840\,\delta^6-468\,\delta^4\,k^2+3213\,\delta^4\,k-4995\,\delta^4-324\,k^2+3321\,k-8487\right)\,f(k-5)\big)$$

$$-\frac{1}{2048}\,\frac{1}{\delta^{10}}\big(\,(k-2)\,(k-3)\,\left(\delta^{14}\,k^2-10\,\delta^{14}\,k+24\,\delta^{14}+299\,\delta^{12}\,k^2-2318\,\delta^{12}\,k+4440\,\delta^{12}+3437\,\delta^{10}\,k^2-22402\,\delta^{10}\,k+36464\,\delta^{10}+2391\,\delta^8\,k^2-16614\,\delta^8\,k+28656\,\delta^8+1827\,\delta^6\,k^2-15390\,\delta^6\,k+31704\,\delta^6-2511\,\delta^4\,k^2+15606\,\delta^4\,k-23592\,\delta^4-81\,\delta^2\,k^2+810\,\delta^2\,k-2016\,\delta^2-243\,k^2+2430\,k-6048\right)\,f(k-4)\big)-\frac{1}{128}\,\frac{1}{\delta^6}\big(\,(k-2)\,\left(18\,\delta^8\,k^2-123\,\delta^8\,k+201\,\delta^8+596\,\delta^6\,k^2-2956\,\delta^6\,k+3704\,\delta^6+144\,\delta^4\,k^2-834\,\delta^4\,k+1170\,\delta^4+108\,\delta^2\,k^2-828\,\delta^2\,k+1488\,\delta^2-162\,k^2+837\,k-1059\right)\,f(k-3)\big)-\frac{1}{64}\,\frac{1}{\delta^6}\big(\,\left(3\,\delta^8\,k^2-18\,\delta^8\,k+24\,\delta^8+398\,\delta^6\,k^2-1418\,\delta^6\,k+1296\,\delta^6+24\,\delta^4\,k^2-114\,\delta^4\,k+120\,\delta^4+18\,\delta^2\,k^2-126\,\delta^2\,k+192\,\delta^2-27\,k^2+108\,k-96\right)\,f(k-2)\big)-f(k-1)\,(4\,k-5)$$

```
> # The functions h_d appearing in the RHS of the recurrence formula are Laurent polynomials in delta. Their degree in 1/\delta are
seq(degree(subs(delta=1/X,coeff(RHS,f(k-i),1)),X),i=1..6);
```

$$0,\ 6,\ 6,\ 10,\ 10,\ 10 \qquad \textbf{(4.1.5}$$

```
> # This implies that these coefficients satisfy Hypothesis (i) with beta=3/4
beta:=3/4;

# The constants a_d(k) are then obtained as the limit of h_d(t,k) (1-256/27 t)^(beta d) when delta->0
seq(factor(limit(coeff(RHS,f(k-d),1)*(increment)^(beta*d),delta=0)),d=1..6);

# These are the values given in the paper.
```

$$\beta := \frac{3}{4}$$

$$0,\ \frac{1}{96}\,\sqrt{6}\,(3\,k-8)\,(3\,k-4),\ 0,\ 0,\ 0,\ 0 \qquad \textbf{(4.1.6}$$

```
> ##### We have obtained the recurrence formula, we now check the initial conditions.
> # We start with the function f^{(0)} for which we have an explicit equation:
map(factor,[solve(factor(subs(s=1,eqh_deltas)),h)]);
# We keep the only solution which has the correct expansion:
```

```
val_f0_delta:=64/(delta^2+3)^3;
series(subs(delta=RootOf(eq_delta,delta),%),t=0);

# (while the other one has not!)
series(subs(delta=RootOf(eq_delta,delta),-64*(delta^2+1)^2/((delta-1)*(delta+1)*(delta^2+3)
^3)),t=0);
```

$$\left[\frac{64}{\left(\delta^2+3\right)^3}, \frac{64}{\left(\delta^2+3\right)^3}, -\frac{64\left(\delta^2+1\right)^2}{(\delta-1)(\delta+1)\left(\delta^2+3\right)^3}\right]$$

$$val\_f0\_delta := \frac{64}{\left(\delta^2+3\right)^3}$$

$$1+3t+15t^2+91t^3+612t^4+4389t^5+O\left(t^6\right)$$

$$t^{-1}-4-8t-36t^2-208t^3+O\left(t^4\right) \quad \textbf{(4.1.7}$$

```
> # We check the singular expansion of f^{(0)}
series(val_f0_delta,delta=0);
# The first singular term is at order delta^2 which corresponds to alpha=1/2
alpha:=1/2;
# In passing we get the value of the constant c_0

limit( (val_f0_delta-64/27)/increment^alpha, delta=0);

valc0:=simplify(%);
```

$$\frac{64}{27}-\frac{64}{27}\delta^2+\frac{128}{81}\delta^4+O\left(\delta^6\right)$$

$$\alpha := \frac{1}{2}$$

$$-\frac{32}{81}\sqrt{2}\sqrt{27}$$

$$valc0 := -\frac{32}{27}\sqrt{2}\sqrt{3} \quad \textbf{(4.1.8}$$

```
> # The value of l_0 is
6+max(floor(alpha/beta), -1);
# so we need to compute 6 more initial conditions.
```

$$6 \quad (4.1.9)$$

```
> # Now the derivatives at s=1 of h can be obtained by a direct Taylor expansion of the algebraic equation. This equation appears to have three branches
algcurves[puiseux](RootOf(eqh_deltas,h),s=1,2):
# which we factor nicely term by term:
branchs:=[seq(map(factor,%[i]),i=1..3)];
```

$$branchs := \left[\frac{64}{(\delta^2+3)^3} - \frac{16\,(s-1)\,(\delta+3)\,(\delta+1)}{(\delta^2+3)^3\,\delta}, \frac{64}{(\delta^2+3)^3} + \frac{16\,(s-1)\,(\delta-1)\,(\delta-3)}{(\delta^2+3)^3\,\delta}, \frac{64\,(\delta^2+1)^2\,(s-1)}{(\delta-1)\,(\delta+1)\,(\delta^2+3)^3} - \frac{64\,(\delta^2+1)^2}{(\delta-1)\,(\delta+1)\,(\delta^2+3)^3} \right] \quad (4.1.10)$$

```
> # To choose which branch is correct, we look at the coefficient of (s-1), which is our candidate for the function f^{(1)},
# and we look which branch gives us a combinatorial (positive!) expansion in t:
series(subs(delta=RootOf(eq_delta,delta),coeff(branchs[1],s-1,1)),t=0);
series(subs(delta=RootOf(eq_delta,delta),coeff(branchs[2],s-1,1)),t=0);
series(subs(delta=RootOf(eq_delta,delta),coeff(branchs[3],s-1,1)),t=0);
# -> so the first branch is the correct one!
```

$$t + 10\,t^2 + 89\,t^3 + 778\,t^4 + 6803\,t^5 + \mathrm{O}(t^6)$$

$$-2 - 7\,t - 40\,t^2 - 271\,t^3 - 2002\,t^4 - 15581\,t^5 + \mathrm{O}(t^6)$$

$$-t^{-1} + 4 + 8\,t + 36\,t^2 + 208\,t^3 + \mathrm{O}(t^4) \quad (4.1.11)$$

```
> # Now that we know that this branch is the correct one, a further Taylor expansion gives us as many derivatives as we want:
map(factor,algcurves[puiseux](RootOf(eqh_deltas,h),s=1,8)[3]):
# Here are the first few ones, they are rational functions of delta:
seq(coeff(%,(s-1),i)*i!,i=1..3);


# We can now directly obtain the values of the constants c_1,...,c_6 by looking at the delta->0 behaviour of these quantities:

seq(limit(coeff(%%,(s-1),i)*i!*increment^(-alpha+beta*i),delta=0, right),i=1..6);
```

```
# Here are the values of c_1,...,c_6
valc:=map(simplify,[%]);
```

$$\frac{16\,(\delta-1)\,(\delta-3)}{(\delta^2+3)^3\,\delta},\ \frac{4\,(\delta^4-10\,\delta^3+16\,\delta^2+6\,\delta+3)\,(\delta-1)^2}{\delta^4\,(\delta^2+3)^3},$$

$$\frac{3}{4}\,\frac{(\delta^7-29\,\delta^6+153\,\delta^5-125\,\delta^4-141\,\delta^3-119\,\delta^2-45\,\delta-15)\,(\delta-1)^3}{\delta^7\,(\delta^2+3)^3}$$

$$\frac{16}{243}\,18^{1/4}\,27^{3/4},\ \frac{8}{27},\ \frac{5}{486}\,18^{3/4}\,27^{1/4},\ \frac{8}{243}\,\sqrt{2}\,\sqrt{27},\ \frac{385}{23328}\,18^{1/4}\,27^{3/4},\ \frac{70}{81}$$

$$valc:=\left[\frac{16}{27}\,3^{3/4}\,2^{1/4},\ \frac{8}{27},\ \frac{5}{54}\,3^{1/4}\,2^{3/4},\ \frac{8}{81}\,\sqrt{2}\,\sqrt{3},\ \frac{385}{2592}\,3^{3/4}\,2^{1/4},\ \frac{70}{81}\right] \quad (4.1.12$$

```
> # At this stage we have verified all hypotheses to apply Theorem 1.8
> # We can be lazy and have Maple solve the recurrence relation for us:
c(k)=(1/96)*sqrt(6)*(3*k-4)*(3*k-8)*c(k-2):
rsolve(%,c(k)):
solrec:=subs(c(1)=valc[1], c(0)= valc0,%);

# The formula is in fact the same for odd and even k:
solrec assuming k::even:
solrec assuming k::odd:
simplify(%%/%);


# And we finally get the value of c_k claimed in the paper...
solrec:=simplify(solrec assuming k::even);

# ...provided we check that it gives the correct values ALSO for k<=6, which is the case:
seq(simplify(solrec)/valc[k], k=1..6);
```

$$solrec := \begin{cases} \frac{32}{27} \frac{6^{-1+\frac{1}{4}k}\,\Gamma\left(\frac{1}{2}k+\frac{1}{3}\right)\Gamma\left(\frac{1}{2}k-\frac{1}{3}\right)3^{\frac{1}{2}+\frac{1}{2}k}\sqrt{2}\sqrt{3}\,8^{-\frac{1}{2}k}}{\pi} & k::even \\ \frac{8}{27} \frac{\Gamma\left(\frac{1}{2}k-\frac{1}{3}\right)\Gamma\left(\frac{1}{2}k+\frac{1}{3}\right)3^{3/4}\,2^{1/4}\,3^{\frac{1}{2}k-\frac{1}{2}}\,8^{-\frac{1}{2}k+\frac{1}{2}}\,6^{\frac{1}{4}k-\frac{1}{4}}}{\pi} & k::odd \end{cases}$$

1

$$solrec := \frac{16}{27} \frac{\Gamma\left(\frac{1}{2}k+\frac{1}{3}\right)\Gamma\left(\frac{1}{2}k-\frac{1}{3}\right)\sqrt{2}\,4^{-k}\,6^{\frac{3}{4}k}}{\pi}$$

$$1, 1, 1, 1, 1, 1 \qquad (4.1.13)$$

```
> # Now applying Theorem 1.8 we find the limiting quantity for the k-th momment:
limit_kth_moment:=solrec/valc0*GAMMA(-alpha)/GAMMA(beta*k-alpha);
```

$$limit\_kth\_moment := \frac{1}{3} \frac{\Gamma\left(\frac{1}{2}k+\frac{1}{3}\right)\Gamma\left(\frac{1}{2}k-\frac{1}{3}\right)4^{-k}\,6^{\frac{3}{4}k}\sqrt{3}}{\sqrt{\pi}\,\Gamma\left(\frac{3}{4}k-\frac{1}{2}\right)} \qquad (4.1.14)$$

```
> # This is not the same value as the one given in Theorem 1.1, but it is in fact equal up to
applying the duplication formula for the GAMMA function.
# Let us have Maple check this:

limit_kth_moment_formulaGivenInTheorem:=(1/2)*sqrt(3)*((1/2)*2^(3/4))^k*GAMMA((1/4)*k+1/3)*
GAMMA((1/4)*k+2/3)/(sqrt(Pi)*GAMMA(1/2+(1/4)*k));
simplify(limit_kth_moment/limit_kth_moment_formulaGivenInTheorem);


# This ends the proof Thm 1.1, namely Equation (3) !!!
# (note that (2) follows from the Carleman criterion)
```

$$limit\_kth\_moment\_formulaGivenInTheorem := \frac{1}{2} \frac{\sqrt{3} \left(\frac{1}{2} 2^{3/4}\right)^k \Gamma\left(\frac{1}{4} k + \frac{1}{3}\right) \Gamma\left(\frac{1}{4} k + \frac{2}{3}\right)}{\sqrt{\pi}\, \Gamma\left(\frac{1}{2} + \frac{1}{4} k\right)}$$

1 **(4.1.15**

```
> # By curiosity, note (this remark is not included in the paper) that there is yet another expression for the k-th moment: as the ratio of two GAMMA functions and not three
  # (but it is less convenient to recognize the limit law)
  limit_kth_moment_otherFormula:=GAMMA(3*k/4)/GAMMA(k/2)*2^(k/4-1)*3^(1-3*k/4);
  simplify(limit_kth_moment/limit_kth_moment_otherFormula);
```

$$limit\_kth\_moment\_otherFormula := \frac{\Gamma\left(\frac{3}{4} k\right) 2^{-1 + \frac{1}{4} k}\, 3^{1 - \frac{3}{4} k}}{\Gamma\left(\frac{1}{2} k\right)}$$

1 **(4.1.16**

Proof of Thm 1.2

```
> # We start by rewriting the equation for G(1,1) in terms of the variable delta instead of z:
> eqg11_deltaw:=factor(numer(subs(z=valz_delta,finaleq_g11)));
```

$$eqg11\_deltaw := \delta^{20} g11^3 w + 26 \delta^{18} g11^3 w + 297 \delta^{16} g11^3 w + 32 \delta^{16} g11^2 w^2 + 32 \delta^{16} g11^2 + 1944 \delta^{14} g11^3 w + 576 \delta^{14} g11^2 w^2 + 576 \delta^{14} g11^2 + 7938 \delta^{12} g11^3 w + 4288 \delta^{12} g11^2 w^2 + 256 \delta^{12} g11^2 w - 256 \delta^{12} g11 w^2 + 4288 \delta^{12} g11^2 + 512 \delta^{12} g11 w + 20412 \delta^{10} g11^3 w + 16704 \delta^{10} g11^2 w^2 - 256 \delta^{12} g11 + 4608 \delta^{10} g11^2 w - 4608 \delta^{10} g11 w^2 + 16704 \delta^{10} g11^2 + 1024 \delta^{10} g11 w + 30618 \delta^8 g11^3 w + 34560 \delta^8 g11^2 w^2 - 4608 \delta^{10} g11 + 34560 \delta^8 g11^2 w - 26368 \delta^8 g11 w^2 + 34560 \delta^8 g11^2 - 33280 \delta^8 g11 w + 17496 \delta^6 g11^3 w + 29376 \delta^6 g11^2 w^2 - 26368 \delta^8 g11 + 138240 \delta^6 g11^2 w - 56320 \delta^6 g11 w^2 + 29376 \delta^6 g11^2 - 231424 \delta^6 g11 w - 19683 \delta^4 g11^3 w - 15552 \delta^4 g11^2 w^2 - 56320 \delta^6 g11 + 311040 \delta^4 g11^2 w - 16128 \delta^4 g11 w^2 - 15552 \delta^4 g11^2 - 631296 \delta^4 g11 w - 39366 \delta^2 g11^3 w - 46656 \delta^2 g11^2 w^2 - 16128 \delta^4 g11 + 262144 \delta^4 w + 373248 \delta^2 g11^2 w + 69120 \delta^2 g11 w^2 - 46656 \delta^2 g11^2 - 801792 \delta^2 g11 w - 19683 g11^3 w - 23328 g11^2 w^2 + 69120 \delta^2 g11 + 524288 \delta^2 w + 186624 g11^2 w + 34560 g11 w^2$$

(4.2.1

$$-23328\, g11^2 - 400896\, g11\, w + 34560\, g11 + 262144\, w$$

```
> #### Recurrence formula
  # Since the function G(1,1)_(w=1+r) is algebraic, it is D-finite, i.e. its derivatives at w=1 satisfy a linear recurrence relation
  # which can be computed automatically by the (great) package gfun
> eqg11_deltaw:
  factor(subs(w=r+1,%)):
  {map(factor,gfun[algeqtodiffeq](%,g11(r)))[1], g11(0)=valfirst0}:
  (gfun[diffeqtorec](%,g11(r),c(k)))[1]:
  subs(seq(c(k-i)=1/(k-i)!*f(k-i),i=0..9),subs(k=k-9,%)):
  # We thus obtain the wanted recurrence to compute the derivatites f(k) (denoted by LaTeX "f^{(k)}" in the paper)
  # f(k)=RHS
  # with
  RHS:=map(factor,collect(expand(solve(%,f(k))),f));
```

$RHS :=$ **(4.2.2**

$$\frac{1}{4096}\,\frac{1}{(\delta^2+3)^7\,\delta^{10}}\Big(f(k-9)\,(k-2)\,(k-3)\,(k-4)\,(k-5)\,(k-6)\,(k-7)\,(k-8)\,(k-9)\,(k-10)\,(\delta-1)^3\,(\delta+1)^3\,(\delta^2+1)^3\,(\delta^4+10\,\delta^2+5)^4\Big) - \frac{1}{8192}\,\frac{1}{(\delta^2+3)^7\,\delta^{10}}\Big(f(k-8)\,(k-2)\,(k-3)\,(k-4)\,(k-5)\,(k-6)\,(k-7)\,(k-9)\,(\delta^2+1)\,(\delta-1)^2\,(\delta+1)^2\,(\delta^4+10\,\delta^2+5)^2\,(9\,\delta^{14}-336\,\delta^{12}\,k+2757\,\delta^{12}-2088\,\delta^{10}\,k+17349\,\delta^{10}-3688\,\delta^8\,k+31969\,\delta^8-1680\,\delta^6\,k+18219\,\delta^6-160\,\delta^4\,k+6463\,\delta^4-200\,\delta^2\,k+4295\,\delta^2-40\,k+859)\Big) - \frac{1}{8192}\,\frac{1}{(\delta^2+3)^6\,\delta^{10}}\Big(f(k-7)\,(k-5)\,(k-6)\,(k-2)\,(k-3)\,(k-4)\,(\delta-1)\,(\delta+1)\,(8\,\delta^{24}\,k^2-107\,\delta^{24}\,k+357\,\delta^{24}-272\,\delta^{22}\,k^2+5318\,\delta^{22}\,k-24762\,\delta^{22}-22768\,\delta^{20}\,k^2+361708\,\delta^{20}\,k-1431540\,\delta^{20}-191248\,\delta^{18}\,k^2+3052882\,\delta^{18}\,k-12139278\,\delta^{18}-548968\,\delta^{16}\,k^2+9099625\,\delta^{16}\,k-37407495\,\delta^{16}-614176\,\delta^{14}\,k^2+11344732\,\delta^{14}\,k-50612772\,\delta^{14}-101024\,\delta^{12}\,k^2+4385120\,\delta^{12}\,k-26998368\,\delta^{12}+775264\,\delta^{10}\,k^2-9662140\,\delta^{10}\,k+29673732\,\delta^{10}+1723704\,\delta^8\,k^2-25342857\,\delta^8\,k+94199271\,\delta^8+1828400\,\delta^6\,k^2-27507650\,\delta^6\,k+104217150\,\delta^6+1020816\,\delta^4\,k^2-15380172\,\delta^4\,k+58240212\,\delta^4+290992\,\delta^2\,k^2-4377334\,\delta^2\,k+16536234\,\delta^2$$

$$+33576 k^2-505077 k+1908027)\Big)-\frac{1}{4096}\frac{1}{(\delta^2+3)^6\delta^{10}}\Big(f(k-6)(k-5)(k-2)(k-3)(k-4)\big(7\delta^{26}k^2-88\delta^{26}k$$
$$+276\delta^{26}+427\delta^{24}k^2-4360\delta^{24}k+10356\delta^{24}-13774\delta^{22}k^2+216136\delta^{22}k-822864\delta^{22}-110590\delta^{20}k^2+1804984\delta^{20}k$$
$$-7058592\delta^{20}-189875\delta^{18}k^2+4060064\delta^{18}k-18291420\delta^{18}+325217\delta^{16}k^2-994304\delta^{16}k-6801516\delta^{16}+1859276\delta^{14}k^2$$
$$-21748976\delta^{14}k+63804288\delta^{14}+4108428\delta^{12}k^2-54754320\delta^{12}k+182748768\delta^{12}+4483697\delta^{10}k^2-61615256\delta^{10}k$$
$$+209969052\delta^{10}+1030653\delta^8k^2-15713160\delta^8k+56951004\delta^8-1483166\delta^6k^2+18560552\delta^6k-58938672\delta^6$$
$$-1201454\delta^4k^2+15621080\delta^4k-51082944\delta^4-371749\delta^2k^2+4879312\delta^2k-16070100\delta^2-48489k^2+636432k$$
$$-2096100\big)\Big)-\frac{1}{4096}\frac{1}{(\delta^2+3)^4\delta^{10}}\Big(f(k-5)(k-2)(k-3)(k-4)\big(6\delta^{22}k^2-69\delta^{22}k+195\delta^{22}+1098\delta^{20}k^2$$
$$-11187\delta^{20}k+28485\delta^{20}-1574\delta^{18}k^2+89045\delta^{18}k-440211\delta^{18}+74454\delta^{16}k^2-380733\delta^{16}k-110229\delta^{16}+379772\delta^{14}k^2$$
$$-3321842\delta^{14}k+6947742\delta^{14}+1050276\delta^{12}k^2-11198142\delta^{12}k+29858226\delta^{12}+1497396\delta^{10}k^2-16984950\delta^{10}k$$
$$+47753466\delta^{10}+501804\delta^8k^2-6347226\delta^8k+19286070\delta^8-160866\delta^6k^2+1395399\delta^6k-2995281\delta^6-153870\delta^4k^2$$
$$+1664865\delta^4k-4541319\delta^4-37422\delta^2k^2+430353\delta^2k-1238391\delta^2-5346k^2+61479k-176913\big)\Big)$$
$$-\frac{1}{2048}\frac{1}{(\delta^2+3)^4\delta^{10}}\Big(f(k-4)(k-2)(k-3)\big(\delta^{22}k^2-10\delta^{22}k+24\delta^{22}+631\delta^{20}k^2-5686\delta^{20}k+12648\delta^{20}$$
$$+15567\delta^{18}k^2-93366\delta^{18}k+109728\delta^{18}+154873\delta^{16}k^2-1059034\delta^{16}k+1676832\delta^{16}+672874\delta^{14}k^2-5247940\delta^{14}k$$
$$+10088304\delta^{14}+1544134\delta^{12}k^2-13242748\delta^{12}k+28237392\delta^{12}+1630302\delta^{10}k^2-14866572\delta^{10}k+33482880\delta^{10}$$
$$+328242\delta^8k^2-3712212\delta^8k+9713856\delta^8-83259\delta^6k^2+415086\delta^6k-271944\delta^6-61933\delta^4k^2+516754\delta^4k-1083384\delta^4$$
$$-6237\delta^2k^2+62370\delta^2k-155232\delta^2-891k^2+8910k-22176\big)\Big)-\frac{1}{128}\frac{1}{(\delta^2+3)^2\delta^6}\Big(f(k-3)(k-2)\big(24\delta^{12}k^2$$
$$-183\delta^{12}k+333\delta^{12}+1472\delta^{10}k^2-8018\delta^{10}k+10350\delta^{10}+7800\delta^8k^2-49437\delta^8k+77607\delta^8+14080\delta^6k^2-94732\delta^6k$$
$$+157428\delta^6+1608\delta^4k^2-14529\delta^4k+29547\delta^4-192\delta^2k^2-258\delta^2k+3006\delta^2-216k^2+1269k-1791\big)\Big)$$

$$-\frac{1}{64}\frac{1}{(\delta^2+3)^2\delta^6}\Big(f(k-2)\,(3\,\delta^{12}k^2-18\,\delta^{12}k+24\,\delta^{12}+664\,\delta^{10}k^2-2854\,\delta^{10}k+2976\,\delta^{10}+3855\,\delta^8k^2-17664\,\delta^8k$$
$$+19824\,\delta^8+6080\,\delta^6k^2-28628\,\delta^6k+32928\,\delta^6+201\,\delta^4k^2-1530\,\delta^4k+2328\,\delta^4-24\,\delta^2k^2-102\,\delta^2k+384\,\delta^2-27\,k^2+108\,k$$
$$-96)\Big)-f(k-1)\,(5\,k-9)$$

```
> # This recurrence has order 9.
  # The functions h_d appearing in the RHS of the recurrence formula are rational functions of
  delta. The order of their poles in 1/\delta are
  seq(degree(   convert(asympt(subs(delta=1/X,coeff(RHS,f(k-i),1)),X,4),polynom),X),i=1..9);
```

$$0, 6, 6, 10, 10, 10, 10, 10, 10 \quad (4.2.3$$

```
> # This implies that these coefficients satisfy Hypothesis (i) with beta=3/4
  beta:=3/4;

  # The constants a_d(k) are then obtained as the limit of h_d(t,k) (1-256/27 t)^(beta d) when
  delta->0
  seq(factor(limit(coeff(RHS,f(k-d),1)*(increment)^(beta*d),delta=0)),d=1..9);

  # In passing, note that the pole at delta=-3 does not provide any new dominant singularity.
```

$$\beta := \frac{3}{4}$$

$$0, \frac{1}{864}\sqrt{6}\,(3\,k-4)\,(3\,k-8), 0, 0, 0, 0, 0, 0, 0 \quad (4.2.4$$

```
> ##### We have obtained the recurrence formula, we now check the initial conditions.
> # We start with the function f^{(0)} for which we have already done the job in the previous
  section (this is the same function)
  map(factor,[solve(factor(subs(s=1,eqh_deltas)),h)]);
  # We keep the only solution which has the correct expansion:
  val_f0_delta:=64/(delta^2+3)^3;
  series(subs(delta=RootOf(eq_delta,delta),%),t=0);

  # (while the other one has not!)
  series(subs(delta=RootOf(eq_delta,delta),-64*(delta^2+1)^2/((delta-1)*(delta+1)*(delta^2+3)
  ^3)),t=0);
```

$$\left[\frac{64}{(\delta^2+3)^3},\ \frac{64}{(\delta^2+3)^3},\ -\frac{64\,(\delta^2+1)^2}{(\delta-1)\,(\delta+1)\,(\delta^2+3)^3} \right]$$

$$val\_f0\_delta := \frac{64}{(\delta^2+3)^3}$$

$$1+3\,t+15\,t^2+91\,t^3+612\,t^4+4389\,t^5+O(t^6)$$

$$t^{-1}-4-8\,t-36\,t^2-208\,t^3+O(t^4) \quad \textbf{(4.2.5}$$

```
> # We check the singular expansion of f^{(0)}
  series(val_f0_delta,delta=0);
  # The first singular term is at order delta^2 which corresponds to alpha=1/2
  alpha:=1/2;
  # In passing we get the value of the constant c_0

  limit( (val_f0_delta-64/27)/increment^alpha, delta=0);

  valc0:=simplify(%);
```

$$\frac{64}{27}-\frac{64}{27}\,\delta^2+\frac{128}{81}\,\delta^4+O(\delta^6)$$

$$\alpha := \frac{1}{2}$$

$$-\frac{32}{81}\,\sqrt{2}\,\sqrt{27}$$

$$valc0 := -\frac{32}{27}\,\sqrt{2}\,\sqrt{3} \quad \textbf{(4.2.6}$$

```
> # The value of l_0 is
  9+max(floor(alpha/beta), -1);
  # so we need to compute 9 more initial conditions.
```

$$9 \quad \textbf{(4.2.7}$$

```
> # Now the derivatives at w=1 can be obtained by a direct Taylor expansion of the algebraic
  equation. This equation appears to have three branches
  (algcurves[puiseux](RootOf(eqg11_deltaw,g11),w=1,3)):
  # which we factor nicely term by term:
```

```
branchs:=[seq(map(factor,%[i]),i=1..3)];
```

$$branchs := \left[-\frac{4\left(3\delta^2+1\right)^2\left(\delta^2+1\right)^2(w-1)^2}{\left(\delta^2+3\right)^4\delta^4} - \frac{64\left(\delta^2+1\right)^2}{(\delta-1)(\delta+1)\left(\delta^2+3\right)^3}, \frac{2(\delta-1)\left(\delta^5-5\delta^4-2\delta^3-10\delta^2+\delta-1\right)(\delta+1)^2(w-1)^2}{\left(\delta^2+3\right)^4\delta^4} + \frac{16(w-1)(\delta-1)(\delta+1)}{\delta\left(\delta^2+3\right)^3} + \frac{64}{\left(\delta^2+3\right)^3}, \frac{2(\delta+1)\left(\delta^5+5\delta^4-2\delta^3+10\delta^2+\delta+1\right)(\delta-1)^2(w-1)^2}{\left(\delta^2+3\right)^4\delta^4} - \frac{16(w-1)(\delta-1)(\delta+1)}{\delta\left(\delta^2+3\right)^3} + \frac{64}{\left(\delta^2+3\right)^3} \right] \quad (4.2.8)$$

```
> # To choose which branch is correct, we look at the coefficient of (w-1), which is our candidate for the function f^{(1)}, and we look which branch gives us a combinatorial (positive!) expansion in t:
valfirst:=factor(coeff(branchs[2],w-1,1));
subs(delta=RootOf(eq_delta,delta),valfirst):
factor(simplify(series(%,t=0,5)));

subs(w=1,diff(devG11t,w));
# -> so the third branch is the correct one!

valfirst0:=factor(coeff(branchs[2],w-1,0));
```

$$valfirst := \frac{16(\delta-1)(\delta+1)}{\delta\left(\delta^2+3\right)^3}$$

$$t+8t^2+63t^3+508t^4+O\left(t^5\right)$$

$$t+8t^2+63t^3+508t^4+4189t^5+35172t^6+299581t^7+O\left(t^8\right)$$

$$valfirst0 := \frac{64}{\left(\delta^2+3\right)^3} \quad (4.2.9)$$

```
> # Now that we know that this branch is the correct one, a further Taylor expansion gives us as many derivatives as we want:
map(factor,algcurves[puiseux](RootOf(eqg11_deltaw,g11),w=1,10)[2]):
```

```
# Quick check that Maple has not changed the order of the branches...
coeff(%,(w-1),1)/valfirst;

# As before we now directly obtain the values of the constants c_1,...,c_9 by looking at the
delta->0 behaviour of these quantities:

seq(limit(coeff(%%,(w-1),i)*i!*increment^(-alpha+beta*i),delta=0, left),i=1..9);

# Here are the values of c_1,...,c_9
valc:=map(simplify,[%]);
```

$$1$$

$$\frac{16}{729}\,18^{1/4}\,27^{3/4},\ \frac{8}{243},\ \frac{5}{13122}\,18^{3/4}\,27^{1/4},\ \frac{8}{19683}\,\sqrt{2}\,\sqrt{27},\ \frac{385}{5668704}\,18^{1/4}\,27^{3/4},\ \frac{70}{59049},\ \frac{85085}{1632586752}\,18^{3/4}\,27^{1/4},\ \frac{700}{4782969}\,\sqrt{2}\,\sqrt{27},\ \frac{37182145}{705277476864}\,18^{1/4}\,27^{3/4}$$

$$valc := \left[\frac{16}{81}\,3^{3/4}\,2^{1/4},\ \frac{8}{243},\ \frac{5}{1458}\,3^{1/4}\,2^{3/4},\ \frac{8}{6561}\,\sqrt{2}\,\sqrt{3},\ \frac{385}{629856}\,3^{3/4}\,2^{1/4},\ \frac{70}{59049},\ \frac{85085}{181398528}\,3^{1/4}\,2^{3/4},\ \frac{700}{1594323}\,\sqrt{2}\,\sqrt{3},\ \frac{37182145}{78364164096}\,3^{3/4}\,2^{1/4}\right] \quad (4.2.10)$$

```
> # We can be lazy and have Maple solve the recurrence relation for us:
c(k)=(1/864)*sqrt(6)*(3*k-4)*(3*k-8)*c(k-2):
rsolve(%,c(k)):
solrec2:=subs(c(1)=valc[1], c(0)= valc0,%);

# The formula is in fact the same for odd and even k:
solrec2 assuming k::even:
solrec2 assuming k::odd:
simplify(%%/%);


# And we finally get the value of c_k...
solrec2:=factor(simplify(solrec2 assuming k::even));

# ...provided we check that it gives the correct values ALSO for k<=9, which is the case:
seq(simplify(solrec2)/valc[k], k=1..9);
```

$$solrec2 := \begin{cases} \dfrac{32}{9}\,\dfrac{6^{-1+\frac{1}{4}k}\,\Gamma\left(\frac{1}{2}k+\frac{1}{3}\right)\Gamma\left(\frac{1}{2}k-\frac{1}{3}\right)\sqrt{2}\,24^{-\frac{1}{2}k}}{\pi} & k::even \\ \dfrac{8}{81}\,\dfrac{\Gamma\left(\frac{1}{2}k-\frac{1}{3}\right)\Gamma\left(\frac{1}{2}k+\frac{1}{3}\right)3^{3/4}\,2^{1/4}\,24^{-\frac{1}{2}k+\frac{1}{2}}\,6^{\frac{1}{4}k-\frac{1}{4}}}{\pi} & k::odd \end{cases}$$

$$1$$

$$solrec2 := \frac{16}{27}\,\frac{\Gamma\left(\frac{1}{2}k+\frac{1}{3}\right)\Gamma\left(\frac{1}{2}k-\frac{1}{3}\right)2^{\frac{1}{2}-k}\,6^{-\frac{1}{4}k}}{\pi}$$

$$1, 1, 1, 1, 1, 1, 1, 1, 1 \qquad (4.2.11$$

```
> # We can compare this value with the one obtained in the previous section (upper path), and the ratio in indeed (1/3)^k.
simplify(solrec2/solrec);


# This ends the proof of Theorem 1.2 !
```

$$3^{-k} \qquad (4.2.12$$

Section 5.3: proof of Theorem 1.3

```
> #### We have obtained an explicit algebraic equation (of degree 5) for the series of intervals with a marked j\in [n]
# with a weight w^(\tildeQ(j)-3\tildeP(j))
final_eq_m11:
> # As always we prefer to work with the variable delta:
delta^2=1-4*z;
valz_delta:=solve(%,z):
eqm11_deltaw:=factor(numer(subs(z=valz_delta,final_eq_m11))):
```

$$\delta^2 = 1 - 4z \qquad (4.3.1$$

```
> # to obtain the first moments we only need to look at the first few derivates of G(1,1)
  algcurves[puiseux](RootOf(eqm11_deltaw, m11), w=1,3):
  expan:=map((x->map(factor,x)),%)[1];

  # Let us check that we have the correct branch:
  eliminate({f0=coeff(expan,w-1,0), eq_delta},delta)[2,1]:
  series(RootOf(%,f0),t=0,10);
  eliminate({f1=coeff(expan,w-1,1), eq_delta},delta)[2,1]:
  series(RootOf(%,f1),t=0,10);
```

$$expan := \frac{4\left(3\,\delta^4-14\,\delta^3+4\,\delta^2-18\,\delta+9\right)\left(\delta+1\right)^2\left(w-1\right)^2}{\delta^2\left(\delta^2+3\right)^4}-\frac{16\left(w-1\right)\left(\delta+1\right)^2}{\left(\delta^2+3\right)^3}-\frac{16\left(\delta-1\right)\left(\delta+1\right)}{\left(\delta^2+3\right)^3}$$

$$t+6\,t^2+39\,t^3+272\,t^4+1995\,t^5+15180\,t^6+118755\,t^7+949344\,t^8+\mathrm{O}\left(t^9\right)$$

$$-t^2-11\,t^3-101\,t^4-887\,t^5-7697\,t^6-66694\,t^7-579270\,t^8-5050141\,t^9+\mathrm{O}\left(t^{10}\right) \quad \textbf{(4.3.2}$$

```
> # Now let us look at singularities
  ff0:=coeff(expan,w-1,0):
  series(%,delta=0,4);

  ff1:=coeff(expan,w-1,1):
  series(%,delta=0,4);

  ff2:=coeff(expan,w-1,2):
  series(%,delta=0,4);
```

$$\frac{16}{27}-\frac{32}{27}\,\delta^2+\mathrm{O}\left(\delta^4\right)$$

$$-\frac{16}{27}-\frac{32}{27}\,\delta+\frac{32}{27}\,\delta^3+\mathrm{O}\left(\delta^4\right)$$

$$\frac{4}{9}\,\delta^{-2}-\frac{140}{81}-\frac{32}{27}\,\delta+\mathrm{O}\left(\delta^2\right) \quad \textbf{(4.3.3}$$

```
> # Conclusion: the singularity of
  # f^(0) is in \delta^2    ~ (1-4t)^(1/2)
  # f^(1) is in \delta^1    ~ (1-4t)^(1/4)
  # f^(2) is in \delta^(-2) ~ (1-4t)^(-1/2)
```

```
# By the transfer theorems this is enough to conclude that
#   E(\tildeQn(J)-3\tildePn(J)) = O(n^(1/4))
# and (more importantly for us)
#    E  [(\tildeQn(J)-3\tildePn(J))^2] = O(n)

# By the Chebyshev inequality the last equation implies the claim (6), and Theorem 1.3 is proved.
```

Details for example 1.11: height a marked point on a Dyck path.

```
> # Let E be the series of Dyck paths, then the series f of Dyck path with a marked abscissa and weight t^size s^height at that abscissa is given by
E=1+t*E^2, f=E^2/(1-s*t*E^2);
# indeed this follows from performing a last passage decomposition to the right and to the left of the path.
```

$$E = E^2 t + 1, f = \frac{E^2}{-E^2 s t + 1} \quad (5.1$$

```
> # Eliminating E we obtain an algebraic equation for f
eqDyck:=eliminate({%},E)[2,1];
```

$$eqDyck := f^2 s^2 t^2 + 2 f^2 s t^2 - f^2 s t + f^2 t^2 + 2 f s t + 2 f t - f + 1 \quad (5.2$$

```
> # And using gfun we get a recurrence relation on coefficients:
factor(subs(s=r+1,eqDyck)):
gfun[algeqtodiffeq](%,f(r)):
gfun[diffeqtorec](%,f(r),c(k))[1]:
subs(seq(c(k-i)=1/(k-i)!*f(k-i),i=0..2),subs(k=k-2,%)):
# We thus obtain the wanted recurrence to compute the derivatites f(k) (denoted by LaTeX "f^{(k)}" in the paper)
# f(k)=RHS
# with
RHS:=map(factor,collect(expand(solve(%,f(k))),f));
```

(5.3

$$RHS := -\frac{(k-1)\,k\,f(k-2)\,t}{-1+4\,t} - f(k-1)\,k \quad \textbf{(5.3}$$

```
> # Therefore we have Hypothesis (i) with L=2 and
  beta:=1/2;
  # and the constants a_d(k) are given by
  a2:=limit(coeff(RHS,f(k-2),1)*(1-4*t),t=1/4);
  a1:=limit(coeff(RHS,f(k-1),1)*sqrt(1-4*t),t=1/4);

  # So the main recurrence is
  rec_ck:= -c(k) + a2*c(k-2);
```

$$\beta := \frac{1}{2}$$

$$a2 := \frac{1}{4}\,(k-1)\,k$$

$$a1 := 0$$

$$rec\_ck := -c(k) + \frac{1}{4}\,(k-1)\,k\,c(k-2) \quad \textbf{(5.4}$$

```
> # We determine c0 and c1 by looking at the expansion. It is simpler to choose \sqrt(1-4t)=R as
  variable
  eliminate({eqDyck, 1-4*t=R^2},t)[2,1];
  dev:=algcurves[puiseux](RootOf(%,f),s=1,2)[2];
  coeff(dev,s-1,0):
  valc0:=limit(%*R,R=0);
  coeff(dev,s-1,1):
  valc1:=limit(%*R^2,R=0);

  # Note that we have
  alpha:=-1/2;
```

$$R^4 f^2 s^2 + 2\,R^4 f^2 s + R^4 f^2 - 2\,R^2 f^2 s^2 - 2\,R^2 f^2 - 8\,R^2 f s + f^2 s^2 - 8\,R^2 f - 2\,f^2 s + f^2 + 8\,f s - 8\,f + 16$$

$$dev := \frac{(-R+1)\,(s-1)}{R^3+R^2} + \frac{2}{R^2+R}$$

$$valc0 := 2$$
$$valc1 := 1$$
$$\alpha := -\frac{1}{2} \quad \textbf{(5.5}$$

```
> # Given c0 and c1 we can now solve the recurrence:
  solrec:=rsolve({rec_ck,c(0)=valc0, c(1)=valc1}, c(k));
```

$$solrec := 2^{1-k}\,\Gamma(k+1) \quad \textbf{(5.6}$$

```
> # And finally apply Theorem 1.4 to get the limit of the k-th moment:
  solrec/valc0*GAMMA(-alpha)/GAMMA(beta*k-alpha);
  # which is the same as this:
  GAMMA(k/2+1);
  simplify(%/%%);

  # This is the k-th moment of a Rayleigh law! (of parameter sigma = 1/sqrt(2))
```

$$\frac{1}{2}\,\frac{2^{1-k}\,\Gamma(k+1)\,\sqrt{\pi}}{\Gamma\left(\frac{1}{2}+\frac{1}{2}k\right)}$$

$$\Gamma\left(\frac{1}{2}k+1\right)$$

$$1 \quad \textbf{(5.7}$$